


**VINCE APPS**

# **40 EDUCATIONAL GAMES FOR THE ORIC ATMOS**







# **40 Educational Games For The Oric Atmos**





# **40 Educational Games For The Oric Atmos**

## **Vince Apps**

**GRANADA**

London Toronto Sydney New York

Granada Technical Books  
Granada Publishing Ltd  
8 Grafton Street, London W1X 3LA

First published in Great Britain by  
Granada Publishing 1984

Copyright © 1984 by Vince Apps

*British Library Cataloguing in Publication Data*

Apps, Vince

40 educational games for the Oric Atmos

1. Electronic games — Juvenile literature

2. Oric Atmos

Juvenile literature

I. Title

794.8'028'5404

GV1469.2

ISBN 0 246 12557 8

Printed and bound in Great Britain by Mackays of Chatham, Kent

All rights reserved. No part of this publication may be reproduced, stored in a retrieval system or transmitted, in any form, or by any means, electronic, mechanical, photocopying, recording or otherwise, without the prior permission of the publishers.



# Contents

Foreword	7
Introduction	8
1 Counties	10
2 Village	16
3 Multiplication and Division	22
4 Addition and Subtraction	25
5 Sum Difference	28
6 Bar Chart	31
7 Masterbrain	34
8 English/French	38
9 Fizz/Buzz	42
10 Highway Code	45
11 Guess the Number	50
12 Word Mix	53
13 Constellations	57
14 History Quiz	60
15 Spelling Test	65
16 Speed Reading	68
17 Area Estimation	71
18 Number Series	75
19 Morse Code	78

20	Areas	83
21	Missing Letter	87
22	Duck Shoot	91
23	Missing Numbers	95
24	Hangman	98
25	Organ	103
26	Anagram	106
27	Compass	109
28	Submarine	113
29	Word Search	117
30	Noughts and Crosses	120
31	Towns	124
32	Vowel Chase	130
33	Parachute	134
34	Racer	139
35	Proverbs	143
36	Simon	147
37	Balance	151
38	Chemist	155
39	Odd Word Out	159
40	Science Quiz	164



# Foreword

This is not a 'how to use your machine' book as we assume that the user will already have a working knowledge of the main commands on the Atmos keyboard. The intention of this collection of programs is to help the young user to become more familiar with their computer, improve their knowledge and have fun at the same time. None of the programs is of great length as we are aware that younger users will want to see the results of their planning as soon as possible.

Virtually all of the programs will have a 'games' element to encourage the user to beat the machine. In some programs the junior programmer will be typing in the answers to the questions which the machine will eventually ask. In order to remove the certainty of a correct answer however, the computer will be requested to make random choices from the information given.

As can be seen from the contents list the programs cover a wide area of interest including mathematics, geography, grammar, weights and measures and music.

The programs have been designed so that they can be adapted to include more complex questions and so increase the user's ability.

# Introduction

This collection of programs has been written for junior users to pit their wits and test their knowledge.

We have designed these programs so you can learn how to use your Atmos computer by typing in your own information. None of the programs is too long to input into your machine as the idea is to help you sharpen up your mind - not to turn you into a typist.

Once the program is in the machine you will be able to begin answering the questions and improving your knowledge. The faster you respond the quicker the computer will ask you the next question.

You will be surprised how quickly you will learn how to use your Atmos, and how soon you will want to move on to the next program.

As soon as you know how to enter these programs - and beat the Atmos - you will be able to change the content in the lists and make things even more difficult for yourself.

## Things to remember

Your Atmos works with it's own language called BASIC. If you try to 'speak' to your machine in another language then nothing will happen, except that you will get a nasty message which says 'error in line'.

Programming is not like writing an essay for class. Your teacher might let you off with a mild caution if you miss out a comma - your Atmos will not.

You have to follow, exactly, the 'characters' shown on the program listings in this book. If you miss a comma, or enter a dash by error, then the program will not work. You cannot put in any other



instruction and expect the machine to work. If you have typed in your program and the Atmos will not 'run' as instructed, then you will have to check your 'list' against the book for 'bugs'. Check carefully before you decide to throw your machine out of the nearest window!

We know that the programs listed in the book are 'bug free' because, not only have we checked and double-checked them, we have reproduced them from our Atmos using a mechanical printer.

We did this because it makes things easier for you and also for our own printers when they produce our books. If you are really having trouble however you can always ask your parents. They **should** be able to help you.

## **Cassette Storage**


Once you have gone through 'inputting' your programs you can store them on cassette tapes for future use.

Cassette tapes will take up some of your pocket money but they mean that you will always have quick access to your list of programs.

Having your programs on tape also means that, when you have improved your programming skills, you will be able to rewrite the programs we have given you.

# 1

## Counties



A few years ago the Government changed the names of most of the counties in England, Wales and Scotland. Some people still call them by their old names so we have come up with this program to help you make sure you know the new names and where they are.

Your computer knows where Gwent is - do you?

### How to play

The computer will draw an outline map for you and give you a list of ten county names to choose from.

The computer will then choose, at random, an area for you to identify. A flashing sign will indicate the area in question. Type in your answers in CAPITALS.

Tries and scores are shown on the screen.

If you are wrong it will be a case of try, try again until you get it right.

## Programming Notes

This program uses the same map drawing routine as 'towns' and 'compass'. See the introduction to 'towns' before typing in 'counties'. The column and row position for the flashing dot and county name is held in the data statements in lines 900 to 990. If your own county is not here, why not try and insert its name and position?

## Program

```

10 REM COUNTIES
20 PAPER 0:INK 6
30 HIRES
50 FOR J=1 TO 10
60 READ M(J,1),M(J,2),T$(J)
70 NEXT J
80 GOTO 140
90 REM DRAW MAP
100 CURSET 12,185,1
110 READ X,Y
111 IF X=-99 THEN RETURN
112 IF X>30 THEN CURSET X,Y,1:GOTO 110
120 DRAW X,Y,1
130 GOTO 110
140 GOSUB 90
160 FOR K=1 TO 10
162 PRINT T$(K):PRINT
164 FOR C=1 TO 10
166 CURSET M(K,1),M(K,2),0:CIRCLE 3,1
174 WAIT 20
176 CURSET M(K,1),M(K,2),0:CIRCLE 3,0
177 WAIT 20
178 NEXT C
180 NEXT K
181 WAIT 200
182 R=INT(RND(1)*10)+1

```

## 12 Counties

```

184 FOR C=1 TO 5
190 CURSET M(R,1),M(R,2),0:CIRCLE 3,0
192 WAIT 10
194 CURSET M(R,1),M(R,2),0:CIRCLE 3,1
196 WAIT 10
198 NEXT C
200 PRINT "TRIES: ";TR
210 PRINT "SCORE: ";SC
220 INPUT "WHICH COUNTY IS THIS";A$
230 CURSET M(R,1),M(R,2),0:CIRCLE 3,0
240 TR=TR+1
250 IF A$=T$(R) THEN 300
260 EXPLODE
270 PRINT "TRY AGAIN!"
274 WAIT 200
280 GOTO 184
300 PING
310 PRINT "CORRECT!"
320 SC=SC+1
334 WAIT 200
340 GOTO 182
900 DATA 76,65,"CLWYD"
910 DATA 30,180,"CORNWALL"
920 DATA 50,122,"DYFED"
930 DATA 82,128,"GWENT"
940 DATA 113,155,"HAMPSHIRE"
950 DATA 132,50,"HUMBERSIDE"
960 DATA 163,147,"KENT"
970 DATA 85,50,"MERSEYSIDE"
980 DATA 165,80,"NORFOLK"
990 DATA 120,38,"YORKSHIRE"
1000 DATA 5,-5,2,2,13,-9,1,-4,2,0,0,-2,
2,0,5,-10,3,-7,5,0,1,-5
1010 DATA 18,-3,12,2,3,-1,2,-5,5,-3,6,-
10,-10,7,-5,1,-3,4
1020 DATA -4,0,-9,-7,-9,0,0,-3,-5,-2,-9
,5,-3,-4,3,0,0,-2,-7,0
1030 DATA 0,-2,3,0,0,-3,-3,0,5,-5,3,0,1
7,-15,2,-5,-2,-10,-2,-4
1040 DATA 1,-2,-10,-5,-3,-2,7,-7,13,-9,
2,2,7,-3,5,3,-3,-4
1050 DATA 6,4,5,-3,-6,0,-5,-7,3,-5,8,-3
,-7,0,0,-8,8,-6,-4,2
1060 DATA 1,-4,-2,2,-2,-2,-5,4,-9,-15,4
,-5,120,21

```

```

1070 DATA 1,2,-2,4,5,-1,4,0,12,13,-3,3,
3,8,4,3
1080 DATA 0,2,-8,-4,8,6,5,6,0,5,-5,6,8,
8,2,-9,6,-2
1090 DATA 7,2,8,4,3,12,-2,14,-9,14,-12,
5,5,0
1100 DATA 0,5,-10,6,5,1,-3,2,6,3,12,-2,
3,1,0,2,-4,1
1110 DATA 2,3,-5,6,-2,3,-4,-1,-10,12,-1
4,-4,-2,3
1120 DATA -10,1,0,1,-10,-4,-10,2,-9,1,2
,2,-12,2,-9,-4
1130 DATA -8,3,-6,3,-3,12,-4,1,-8,-5,-7
,1,-3,4,-1,-2
1140 DATA -2,2,0,4,-4,3,-6,-5,-6,3,-5,-
3,106,167,4,-2
1150 DATA 6,2,-4,4,-5,-4,155,135,4,0,0,
2,-2,0,-2,-2
1160 DATA 53,62,8,-5,-10,-5,-3,3,0,2,3,
6,-99,0

```

```

10 REM COUNTIES
14 REM SCOTLAND & NI
20 PAPER 0:INK 6
30 HIRES
50 FOR J=1 TO 10
60 READ M(J,1),M(J,2),T$(J)
70 NEXT J
80 GOTO 140
90 REM DRAW MAP
100 CURSET 38,172,1
110 READ X,Y
111 IF X=-99 THEN RETURN
112 IF X>30 THEN CURSET X,Y,1:GOTO 110
120 DRAW X,Y,1
130 GOTO 110
140 GOSUB 90
160 FOR K=1 TO 10
162 PRINT T$(K):PRINT
164 FOR C=1 TO 10
166 CURSET M(K,1),M(K,2),0:CIRCLE 3,1
174 WAIT 20

```



# 14 Counties

```

176 CURSET M(K,1),M(K,2),0:CIRCLE 3,0
177 WAIT 20
178 NEXT C
180 NEXT K
181 WAIT 200
182 R=INT(RND(1)*10)+1
184 FOR C=1 TO 5
190 CURSET M(R,1),M(R,2),0:CIRCLE 3,0
192 WAIT 10
194 CURSET M(R,1),M(R,2),0:CIRCLE 3,1
196 WAIT 10
198 NEXT C
200 PRINT "TRIES:";TR
210 PRINT "SCORE:";SC
220 INPUT "WHICH COUNTY IS THIS";A$
230 CURSET M(R,1),M(R,2),0:CIRCLE 3,0
240 TR=TR+1
250 IF A$=T$(R) THEN 300
260 EXPLODE
270 PRINT "TRY AGAIN!"
274 WAIT 200
280 GOTO 184
300 PING
310 PRINT "CORRECT!"
320 SC=SC+1
334 WAIT 200
340 GOTO 182
900 DATA 52,145,"ANTRIM"
910 DATA 162,145,"BORDERS"
920 DATA 60,162,"DOWN"
930 DATA 10,165,"FERMANAGH"
940 DATA 150,102,"FIFE"
950 DATA 165,54,"GRAMPIAN"
960 DATA 120,27,"HIGHLAND"
970 DATA 147,122,"LOTHIAN"
980 DATA 135,88,"TAYSIDE"
990 DATA 23,148,"TYRONE"
1000 DATA -3,1,-4,-2,0,-2,-4,-3,-1,-4,-
2,-2,-2,3
1010 DATA -5,6,-2,3,-3,1,-3,-3,-4,-2,-1
,-3,-4,-1,1,-5,-1,-6,5,-2
1020 DATA 10,-3,-5,-4,0,-2,4,0,4,-4,7,-
10,3,1,3,-3,0,-4,3,2,6,-1
1030 DATA 3,-3,1,2,1,0,1,-1,4,3,5,-2,0,
8,3,0,-2,3,6,9

```

1040 DATA 1,-3,1,0,1,6,-7,8,9,-6,3,5,0,  
3,2,2,0,6,-1,-3,-2,-3

1050 DATA -1,-2,-2,-2,-1,2,2,6,-3,3,5,-  
1,1,3,-3,4

1060 DATA -2,-1,-2,2,-1,-3,-2,-3,-4,6,-  
5,4,-8,-5,-4,2

1070 DATA 150,161,-15,2,0,2,-5,2,-2,3,-  
3,0,-2,-2,-3,2,-4,-7

1080 DATA 0,11,-2,1,-3,-2,-2,-4,-5,-3,-  
3,-4,-4,3,1,6,-1,1,-5,-13

1090 DATA 3,-3,1,4,1,0,0,-4,3,-5,3,-10,  
3,-2,0,-3,-4,-5,1,-10,3,-3

1100 DATA 6,2,0,-1,-5,-10,-3,8,-2,-2,1,  
-6,-2,0,-1,-6,-2,3,-2,-7,8,-7

1110 DATA -5,0,-6,5,0,10,-6,19,-3,0,-1,  
-2,2,-2,0,-5,4,-5,-1,-13

1120 DATA 4,-5,-3,-4,13,-25,-1,0,-13,15,  
-9,-5,0,-1,4,-3

1130 DATA 30,-70,6,3,1,4,6,-4,2,0,-2,3,  
10,-4,4,1,10,1,4,-5

1140 DATA 6,4,-3,6,0,5,-20,15,0,3,-2,0,  
0,2,5,-1,0,2,-5,4

1150 DATA -13,13,1,0,8,-7,1,1,17,-7,1,2,  
6,2,4,-2,10,2,5,-1,6,2


1160 DATA 1,1,0,4,-2,5,-9,23,-5,5,1,2,-  
5,6,-2,1,-4,-2

1170 DATA -8,4,-2,3,0,1,10,-4,-2,4,2,1,  
3,3,-2,3,-8,0,-3,4

1180 DATA -2,1,-14,-4,-2,-1,0,1,11,7,5,  
-1,4,3,2,0,4,-4,1,0,5,4

1190 DATA 6,0,1,1,5,1,3,4,0,4,2,3,-6,8,  
-2,10,-20,14,-99,0

## 2 Village



You have just been appointed the chief of a village of natives whose lives depend on their crops of wheat. If you manage the crops properly then the village will prosper and the population will increase but should you make a mistake then people will starve, people will die and you will be attacked by an angry mob.

By the way, your people need about  $2\frac{1}{2}$  bags of wheat each to survive for a year. Give them less and they starve and they won't like that. If you give them 5 bags each they will be pleased and may forgive you for past mistakes.

Look out as well for the rats which always attack your crops in the warehouse. The more you store, the greater will be the losses from rats.

## How to play

The screen will show you that you are in your first year as leader. You will begin with a certain population and a certain amount of wheat. There are symbols for people and sacks of wheat and each symbol represents ten units. The computer will ask you how much of your wheat you wish to sow. Remember to keep some back as it might be a bad harvest.

The computer will then tell you how much wheat you have grown and you will be asked how much you wish to give your people.

Try and survive ten years as leader.

## Program

```

10 REM VILLAGE
18 PAPER 7:INK 0
20 CLS
30 PRINT CHR$(17)
50 FOR N=0 TO 31
60 READ D
70 POKE 46856+N,D
80 NEXT N
90 DATA 0,12,30,63,63,51,51,51
100 DATA 12,12,30,45,45,12,18,18
110 DATA 0,0,0,30,30,30,30,30
120 DATA 56,8,30,30,30,30,12,12
130 SF=0
140 PP=100
150 WHT=250
160 YR=1
170 ANG=0
180 AE=2.4
190 GOSUB 900
200 GOSUB 1000
210 FOR J=8 TO 14:PRINT @15,J;CHR$(130)
:NEXT J
220 PRINT @18,8;"aaa"
230 PRINT @17,9;"a    a"
240 PRINT @16,10;"a      a"
250 PRINT @16,11;"a      a"
260 PRINT @16,12;"a      a"
270 PRINT @17,13;"a    a"

```

18 Village

```

280 PRINT @18,14;"aaa"
290 PRINT @18,11;CHR$(129);"b";CHR$(130
)
294 GOSUB 300
296 GOTO 330
300 PRINT @2,3;CHR$(147);"POPULATION: "
;PP;" ";CHR$(151)
306 WHT=INT(WHT+0.5)
310 PRINT @21,3;CHR$(147);"WHEAT: ";WHT
;" ";CHR$(151)
320 PRINT @12,0;CHR$(147);"YEAR: ";YR;C
HR$(151)
322 RETURN
330 HVS=INT(RND(1)*3)+1
332 IF HVS=1 THEN A$="POOR"
334 IF HVS=2 THEN A$="FAIR"
336 IF HVS=3 THEN A$="GOOD"
338 M$="THE WITCH-DOCTOR PREDICTS A "+A
$+" HARVEST.":GOSUB 1100
340 PRINT @2,22;"";:INPUT "HOW MUCH SEE
D WILL YOU SOW";SD
342 PRINT @2,22;"
"

344 IF SD<0 THEN EXPLODE:GOTO 340
350 IF SD<=WHT THEN 360
352 EXPLODE:M$="YOU DO NOT HAVE "+STR$(
SD)+" BAGS!":GOSUB 1100:GOTO 340
360 IF SD<=PP THEN 370
362 EXPLODE:M$="THERE ARE NOT ENOUGH PE
OPLE TO SOW "+STR$(SD)+" BAGS!"
364 GOSUB 1100:GOTO 340
370 CRP=INT(RND(1)*2*HVS*SD)
372 WHT=WHT-SD
374 GOSUB 300:GOSUB 1000
376 WAIT 250
380 M$="YOUR CROP WAS"+STR$(CRP)+" BAGS
OF WHEAT.":GOSUB 1100
384 IF HVS=1 OR CRP>1.5*SD THEN 400
386 M$="EVEN WITCH-DOCTORS CAN BE WRONG
!":GOSUB 1100
400 WHT=WHT+CRP
402 GOSUB 300:GOSUB 1000:WAIT 250
404 PRINT @2,22;"";:INPUT "HOW MUCH WHE
AT WILL YOU GIVE TO YOUR PEOPLE";ET
406 PRINT @2,22;"
"

```

```

408 IF ET<0 THEN 404
412 IF ET<WHT THEN 415
413 EXPLODE:M$="YOU DO NOT HAVE "+STR$(
ET)+" BAGS!":GOSUB 1100:PRINT
414 GOTO 404
415 IF ET>PP*AE THEN 417
416 M$="YOUR PEOPLE ARE HUNGRY!":GOSUB
1100:SF=1
417 IF ET<PP*AE*2 THEN 420
418 M$="YOUR PEOPLE ARE HAPPY!":GOSUB 1
100:ANG=ANG-1
420 WHT=WHT-ET
422 GOSUB 300
424 GOSUB 1000
430 FOR J=26 TO 30 STEP 2
440 FOR K=4 TO 10
450 PRINT @J,K;"d"
460 PLAY 1,0,0,0:MUSIC 1,3,9,7:WAIT 5:P
LAY 0,0,0,0
480 PRINT @J,K;" "
490 NEXT K
500 WAIT 50
510 NEXT J
512 RTS=INT(RND(1)*WHT/4)
514 M$="RATS ATE"+STR$(RTS)+" BAGS OF W
HEAT!":GOSUB 1100
516 WHT=WHT-RTS
518 GOSUB 300:GOSUB 1000
520 IF SF=0 THEN 700
530 DD=INT(RND(1)*0.5*(PP*AE-ET))+1
534 IF DD>=PP THEN DD=PP-1
540 M$=STR$(DD)+" OF YOUR PEOPLE HAVE S
TARVED TO DEATH!":GOSUB 1100
544 PP=PP-DD
550 GOSUB 300:GOSUB 900
560 M$="YOUR PEOPLE ARE ANGRY!":GOSUB 1
100
570 ANG=ANG+1
580 IF ANG<>3 THEN 590
582 M$="YOU HAVE LET TOO MANY STARVE. Y
OUR PEOPLE WANT A NEW LEADER!"
584 GOSUB 1100:GOTO 870
590 FOR J=3 TO PP/10+2
600 IF J>10 THEN 660
610 PRINT @J,9;" "
620 PRINT @J,8;"b"

```

```

630 EXPLODE
640 WAIT 10
660 NEXT J
670 PRINT @3,8;"                               ";CHR$(130)
690 GOTO 740
700 PC=INT(RND(1)*WHT*0.5/AE)+1
710 M$=STR$(PC)+" PEOPLE JOINED YOUR VI
LLAGE!":GOSUB 1100
720 PP=PP+PC
730 GOSUB 300
740 GOSUB 900
750 SF=0
760 YR=YR+1
770 M$="ANOTHER YEAR HAS PASSED.":GOSUB
1100
790 IF YR=11 THEN 820
800 GOTO 294
820 PING
840 PING
850 PRINT @2,17;CHR$(150);"WELL DONE! Y
OU HAVE COMPLETED";CHR$(151)
852 PRINT @2,18;CHR$(150);"YOUR 10 YEAR
S IN OFFICE.";CHR$(151)
860 PRINT @2,20;"YOUR SCORE IS ";INT((P
P+WHT/AE)*10):GOTO 1150
870 FOR J=7 TO 11
872 PRINT @J,9;"b"
874 WAIT 5
876 PRINT @J,9;" "
878 NEXT J
880 PRINT @13,10;"b":WAIT 10
882 PRINT @13,10;" "
888 FOR J=1 TO 5
890 PRINT @19,11;" "
891 ZAP
892 WAIT 10
894 PRINT @19,11;"b"
895 WAIT 10
896 NEXT J
898 GOTO 1150
900 P$="b"
910 FOR J=9 TO 15
920 PRINT @2,J;CHR$(132)
930 FOR K=1 TO 10
940 IF PP<K*10+(J-9)*100 THEN P$=" "
950 PRINT @K+3,J;P$

```

```


960 NEXT K
965 PRINT @15,J;CHR$(130)
970 NEXT J
980 RETURN
1000 P$="c"
1010 FOR J=9 TO 15
1020 PRINT @24,J;CHR$(133)
1030 FOR K=1 TO 10
1040 IF WHT<K*10+(J-9)*100 THEN P$=" "
1050 PRINT @K+24,J;P$
1060 NEXT K
1070 NEXT J
1080 RETURN
1100 INK 0:PRINT @2,19;M$
1110 WAIT 500
1120 PRINT @2,19;"
"
1124 PRINT @2,20;"
"
1130 RETURN
1150 PRINT @2,23;"":INPUT "PLAY AGAIN"
;Q$
1160 IF LEFT$(Q$,1)="N" THEN END
1170 CLS
1180 GOTO 130

```



# 3

## Multiplication and Division



The first thing you have to do before you begin to play this brain teaser is to put your calculators away in a cupboard and bring out your pencils and paper. This game is a test of your mental ability and agility. No cheating, now, by using any help.

### How to play

Your computer will begin by asking you if you wish to play multiplication or division.

Type M or D then press RETURN.

You will then be given a simple multiplication question such as 1436 x 26.

If you are correct, you get a nice little tick and the score will be recorded on the board.

If you are wrong, you will receive a large cross against your answer and the correct answer will pop up on the screen.

The program will run for a total of twenty sums and will then give your grand total of correct answers against attempts.

If you wish to change from one type of sum to another you can wait until you have answered your twenty questions then type RUN, or you can press CTRL and C together, then RUN. Either will bring you back to the beginning of the program.

## **Programming Hints**

You can make the sums easier for younger members of the family by altering lines 200 and 210.

## **Program**

```

10 REM MULTIPLICATION AND DIVISION
18 PAPER 7:INK 0
20 CLS
70 FOR N=0 TO 23
80 READ D
90 POKE 46856+N,D
100 NEXT N
110 DATA 0,12,0,63,63,0,12,0
120 DATA 0,0,1,2,4,40,16,0
130 DATA 0,0,0,34,20,8,20,34
140 PRINT:PRINT
150 INPUT "MULTIPLICATION OR DIVISION";
C$
160 IF LEFT$(C$,1)="M" THEN S$="X":GOTO
184
170 IF LEFT$(C$,1)="D" THEN S$="a":GOTO
184


```

## 24 *Multiplication and Division*

```
180 GOTO 150
184 CLS
186 PRINT:PRINT
190 IF S$="a" THEN 260
200 R1=INT(RND(1)*9000)+100
210 R2=INT(RND(1)*100)+1
230 CA=R1*R2
240 PRINT @2,21;R1;" ";S$;" ";R2;" = ";
250 GOTO 310
260 R1=INT(RND(1)*100)+1
270 R2=INT(RND(1)*100)+1
280 CA=R1
290 PRINT @2,21;R1*R2;" ";S$;" ";R2;" =
";
310 PRINT @2,24;"":INPUT ANS:PRINT @2,
24;"
320 PRINT @19,21;ANS;" ";
330 SM=SM+1
340 IF ANS=CA THEN 380
350 EXPLODE
360 PRINT CHR$(129);"c";" ";
364 PRINT CA
370 GOTO 410
380 PING
390 PRINT CHR$(129);"b"
400 SC=SC+1
410 PRINT @2,26;"":PRINT
420 IF SM=20 THEN 460
430 PRINT @5,0;CHR$(147);"SUMS: ";SM;CH
R$(151)
440 PRINT @20,0;CHR$(147);"SCORE: ";SC;
CHR$(151)
446 PRINT @2,1;"
"
450 GOTO 190
460 WAIT 250
470 CLS
480 PRINT @6,8;CHR$(146);"YOU HAD";SC;"
CORRECT ANSWERS";CHR$(151)
490 PRINT @8,10;CHR$(146);"OUT OF 20 QU
ESTIONS.";CHR$(151)
500 END
```

# 4

## Addition and Subtraction



When you cover other sums programs in this book you will find that we recommend that you don't use pencils, paper or calculators because you probably won't have them the next time you are sent out to shop for your parents. It helps to be able to do your own sums in your head because it might save you having to go back to the shops because you didn't have enough money the first time you went.

### How to play

We have made this game look just like it would on your school papers. You even fill in the totals by putting the numbers in from right

to left just like you do in your maths books. You don't have to press RETURN. Your computer will do this for you.

If you get a correct answer a little yellow munchman will run across the screen and eat away the whole sum.

If you are wrong however the munchman will only eat away the incorrect answer and you will have to try again.

Give your munchman a full stomach by getting as many correct as you can because the score board will show how many tries and how many mistakes you made.

### Program

```

10 REM ADDITION AND SUBTRACTION
30 PRINT CHR$(17)
60 PAPER 0: INK 7
64 CLS
70 FOR N=0 TO 15
80 READ D
90 POKE 46856+N,D
100 NEXT N
110 DATA 0,30,63,45,63,45,51,30
120 DATA 0,30,63,45,63,33,45,30
160 A=INT(RND(1)*40000)+10000
170 B=INT(RND(1)*40000)+10000
190 SN=1
200 IF RND(1)>0.48 THEN SN=-1
204 IF SN<0 AND (A-B)<10000 THEN 160
210 PRINT @13,9;A
220 PRINT @13,10;B
250 IF SN<0 THEN PRINT @19,9;"-"
260 IF SN>0 THEN PRINT @19,9;"+"
270 CA=A+B*SN
280 SM=SM+1
290 A$=""
300 PRINT @5,1;CHR$(147);CHR$(128);"SUM
S: ";SM;CHR$(144)
304 PRINT @18,1;CHR$(147);CHR$(128);"SC
ORE: ";SC;CHR$(144)
310 K$=KEY$
320 IF K$="" THEN 310

```


```

330 IF K$>"9" OR K$<"0" THEN 310
344 FOR D=1 TO 20:NEXT D
350 A$=K$+A$
354 PRINT @19-LEN(A$),12;A$
360 IF LEN(A$)<>5 THEN 310
370 IF VAL(A$)=CA THEN 480
380 EXPLODE:WAIT 100
390 PRINT @11,17;CHR$(145);CHR$(128);"T
RY AGAIN!";CHR$(144)
400 FOR J=10 TO 22
410 PRINT @J,12;CHR$(129);"b"
420 MUSIC 1,3,3,0:PLAY 1,0,1,900
430 WAIT 20
440 PRINT @J,12;"  "
450 NEXT J
460 PRINT @11,17;"          "
470 GOTO 280
480 PING:WAIT 100
490 PRINT @12,17;CHR$(147);CHR$(128);"C
ORRECT!";CHR$(144)
500 SC=SC+1
510 FOR J=9 TO 13
520 FOR K=10 TO 22
530 PRINT @K,J;CHR$(131);"a"
540 MUSIC 1,4,1,0:PLAY 1,0,1,300
544 WAIT 5
550 PRINT @K,J;"  "
560 NEXT K
570 NEXT J
580 PRINT @12,17;"          "
590 GOTO 160

```

# 5

## Sum Difference



We thought about calling this game 'some difference' as it isn't nearly as easy as it looks at first.

You are being told by your computer that it is thinking of two numbers between 1 and 20. It will tell you the total sum and the **difference** between the two numbers. All you have to do is correctly guess the answer.

Example: The sum of the numbers is 13  
The difference is 9  
What are the numbers?  
Answer 2 and 11

Simple isn't it?

## How to play

Your computer will tell you the sum of the numbers it is thinking of and the difference and ask for your answers.

After each number press RETURN.

If you are correct the score increases on the top of the board.

As you become better at this game the computer will move the range of numbers from 1 - 20 to 1 - 25 and so on.

A wrong answer will end the game completely and you will be asked if you wish to compete again. High scores will be recorded on the screen to allow you to compete for the high score title.

## Programming Hints

You can make the game tougher from the very beginning by increasing the value in line 50 so that the range is wider immediately.

## Program

```

10 REM SUM DIFFERENCE
40 HS=0
50 MAX=20
60 SC=0
64 PAPER 7:INK 0
70 CLS
80 PRINT @4,1;CHR$(149);CHR$(128);"SCOR
E: ";SC
90 PRINT @17,1;CHR$(149);CHR$(128);"HI
SCORE: ";HS;CHR$(151)
100 R1=INT(RND(1)*MAX)+1
110 R2=INT(RND(1)*MAX)+1
120 IF R2>=R1 THEN 100
130 INK 4:PRINT @2,8;"I AM THINKING OF
TWO NUMBERS BETWEEN 1 AND ";MAX
140 PRINT @2,11;"THE SUM OF THE NUMBERS
IS ";R1+R2

```




### 30 *Sum Difference*

```
150 PRINT @2,13;"THE DIFFERENCE IS ";R1
-R2
154 PRINT:PRINT
160 INPUT "WHAT ARE THE TWO NUMBERS";N1
,N2
170 IF N1=N2 THEN 190
180 IF (N1=R1 OR N1=R2) AND (N2=R1 OR N
2=R2) THEN 270
190 EXPLODE
200 PRINT @2,19;CHR$(147);;"THE TWO NUM
BERS WERE ";R1;" AND ";R2;CHR$(151)
202 PRINT:PRINT
210 PRINT R1;"+";R2;"=";R1+R2
220 PRINT R1;"-";R2;"=";R1-R2
224 PRINT:PRINT
230 PRINT "YOUR SCORE IS ";SC
234 PRINT:PRINT
240 IF SC>HS THEN HS=SC:PRINT "A NEW HI
GH SCORE!":PRINT:PRINT
250 INPUT "PLAY AGAIN";Q$
260 IF LEFT$(Q$,1)="N" THEN END
264 GOTO 50
270 PING
280 PRINT @2,19;"THAT IS CORRECT!"
290 WAIT 300
300 SC=SC+1
310 MAX=MAX+5
320 GOTO 70
```

# 6

## Bar Chart



If you are planning to keep a record of how tall you have grown, or the height of a plant, or the rainfall in your garden, or the temperature changes, or how much pocket money you have been spending on sweets, toys or even computer games then this is for you.

The computer will draw you charts in six different colours so that you can lay one colour over another to show how much, or how little, change there has been since the last time you checked.

### How to play

Your computer will show you an upright (perpendicular) line and a base (horizontal) line and will ask you first to decide the values you wish to use. This means that you have to decide if you want the

## 32 Bar Chart

chart to rise two, five or even ten units at a time. You are then asked which colour you would like to see the chart printed in and you choose between 1 and 5.

You will then be asked to type in the figure you wish the chart to show. For example, if the temperature on the first day had been 24 Centigrade then you type in 24, and for the next day, 22, and so on. At the end of two weeks you will have finished 14 numbers and will see the high and low temperatures for the period covered.

If you wish to break out of the program at any time you type any letter and the program will take you back to the list of colours. You can then choose another colour to lay on top of the first one chosen so that you can see a comparison with last year's temperature figures, if you wish. When you break out the computer will ask you if you want to continue with more values. Type in Y for yes or N for no, then the number for the new colour.

### Program

```
10 REM BAR CHART
18 PAPER 7:INK 0
20 CLS
30 FOR N=0 TO 7
40 READ D
60 NEXT N
70 DATA 31,31,31,31,31,31,31,31
84 IF M<>0 THEN WAIT 400:CLS
88 PRINT @2,10;" ";
90 INPUT "WHAT IS YOUR MAXIMUM VALUE (M
MULTIPLES OF 20)";M
94 IF M<20 THEN PRINT @11,6;CHR$(146);"
TOO SMALL!";CHR$(151):GOTO 84
100 IF M>100 THEN PRINT @11,6;CHR$(149)
;"TOO BIG!";CHR$(151):GOTO 84
101 PRINT:PRINT
102 INPUT "COLOUR (1 TO 6)";C
104 IF C<1 OR C>6 THEN 102
106 C=C+128
108 CLS
110 S=INT(M/20)
116 P=4
120 FOR J=1 TO 19
```


```

124 IF S*J>9 THEN P=3
130 PRINT @P,20-J;S*J
140 NEXT J
146 N=5
150 FOR J=1 TO 15
154 N=N+2
160 PRINT @N,21;RIGHT$(STR$(J),LEN(STR$(J))-1)
170 NEXT J
186 N=6
190 PRINT @3,23;"";:INPUT "VALUE";V$
194 PRINT @3,23;"
200 IF LEFT$(V$,1)>"9" OR LEFT$(V$,1)<"
0" THEN 300
210 VL=VAL(V$)
220 IF VL>M THEN VL=M
230 FOR J=0 TO VL/S
240 IF J=0 THEN 260
250 PRINT @N,20-J;CHR$(C);"a"
260 NEXT J
270 N=N+2
280 IF N>35 THEN 300
290 GOTO 190
300 PRINT @3,24;"";:INPUT "MORE";Q$
310 IF LEFT$(Q$,1)="N" THEN END
320 INPUT "COLOUR";C
330 IF C<1 OR C>6 THEN 320
334 C=C+128
340 PRINT @3,24;"
350 PRINT @2,25;"
360 GOTO 186

```

# 7

## Masterbrain



This is one of the most difficult programs in the book in our opinion as you have to really think about every move you make.

Your computer will choose four coloured pegs and place them in four positions in its memory.

You have to guess what the colours are, how many of each colour and in fact what order the computer has placed them.

You could be sitting here for days so you had better get a biscuit and a drink before you start.

### How to play

At the beginning you will be asked your age as the game has

different levels of difficulty. You could cheat to begin with and say you are only five years old - it might make things easier.

The colours are as follows

Key 3	Red
4	Green
5	Yellow
6	Blue

You are asked for your first guess and you press the four numbers for your guess and RETURN.

When you have entered your guesses the computer will show you a yellow peg if you have guessed a correct colour but put it in the wrong column. You will get a red peg if your guess of colour and column is correct. If you have guessed a colour the computer hasn't chosen you will get a blank column.

Now you may know that your first guess of colours is half right you have to go on to find out which column the colours are really in.

You could start by making all the keys 3 (red) and if the computer showed you a blank screen then you would know that it hadn't used red and you could ignore that number in your next choice. You can go on trying this method but it takes a long time and it will mean that your score is pretty awful when it shows the number of moves you took, at the end of the game.

If at any stage you want to give up, type 'quit' as your guess.

## Program

```

10 REM MASTERBRAIN
30 CLS
40 PAPER 0: INK 7
90 INPUT "HOW OLD ARE YOU"; AGE
91 CLS
92 IF AGE<1 OR AGE>18 THEN MAX=10: GOTO
100
96 MAX=28-AGE

```

```

100 PRINT @3,6;"COLOURS:"
110 PRINT @3,8;CHR$(145);" ";CHR$(144);
120 PRINT "RED (3)"
130 PRINT @3,10;CHR$(146);" ";CHR$(144)
;
140 PRINT "GREEN (4)"
150 PRINT @3,12;CHR$(147);" ";CHR$(144)
;
160 PRINT "YELLOW (5)"
170 PRINT @3,14;CHR$(148);" ";CHR$(144)
;
180 PRINT "BLUE (6)"
200 PRINT @5,1;CHR$(147);CHR$(128);"WIN
S: ";WN;CHR$(144)
210 PRINT @18,1;CHR$(147);CHR$(128);"GA
MES: ";GM;CHR$(144)
220 PRINT @2,18;
240 TR=1
260 FOR J=1 TO 4
270 R(J)=INT(RND(1)*4)+3
280 NEXT J
300 PRINT "GUESS NO.";TR;
310 INPUT A$
314 IF LEN(A$)<>4 THEN 310
316 IF A$="QUIT" THEN 652
320 FOR N=1 TO 4
330 IF MID$(A$,N,1)>"6" OR MID$(A$,N,1)
<"3" THEN 310
340 L(N)=VAL(MID$(A$,N,1))
350 PRINT CHR$(L(N)+142);" ";CHR$(144);
" ";
360 NEXT N
370 PRINT " ";
380 FOR N=1 TO 4
390 S(N)=R(N)
400 NEXT N
420 RD=0:BL=0
430 FOR J=1 TO 4
440 IF L(J)=S(J) THEN BL=BL+1:S(J)=-1:L
(J)=0
450 NEXT J
456 FOR J=1 TO 4
458 IF L(J)=0 THEN 490
460 FOR K=1 TO 4
470 IF L(J)=S(K) THEN RD=RD+1:S(K)=-1:L
(J)=0

```

```


480 NEXT K
490 NEXT J
500 IF BL=0 THEN 540
510 FOR J=1 TO BL
520 PRINT CHR$(129);"I";CHR$(135);
530 NEXT J
540 IF RD=0 THEN 580
550 FOR J=1 TO RD
560 PRINT CHR$(131);"I";CHR$(135);
570 NEXT J
580 PRINT:PRINT
590 IF BL=4 THEN 670
600 TR=TR+1
610 IF TR<>MAX THEN 300
630 PING
640 EXPLODE
650 PRINT "OUT OF GUESSES!"
652 PRINT:PRINT "THE COLOURS WERE: ";:P
RINT
654 FOR N=1 TO 4
656 PRINT CHR$(R(N)+142);" ";CHR$(144);
" ";
658 NEXT N
660 PRINT:PRINT
664 GOTO 700
670 PLAY 1,0,0,0:MUSIC 1,3,8,8:WAIT 50
680 MUSIC 1,4,1,12:WAIT 70:PLAY 0,0,0,0
690 WN=WN+1
692 MAX=MAX-1
694 IF MAX<5 THEN MAX=5
700 GM=GM+1
704 PRINT @2,24;""::PRINT:PRINT
710 INPUT "PLAY AGAIN";Q$
720 CLS
730 IF LEFT$(Q$,1)="N" THEN 750
740 GOTO 100
750 PRINT @8,8;CHR$(146);CHR$(128);"THA
NKS FOR THE GAME! ";CHR$(144)
760 PRINT:PRINT
770 END

```



# 8

## English/French



Imagine you have gone on holiday with your family to France and your Mum and Dad can't remember the French for an hotel. You just walk up behind them and say 'It's an auberge, Dad'.

They will probably be so surprised that you could knock them down with a pain (that's French for a loaf of bread, but of course you will know that.)

### How to play

The computer will concentrate on nouns but you can change the program later to widen your knowledge.

Always remember when answering in French to use le or la before your word, or your computer will tell you that you have made an error. Our computers are very correct things you know.

Keep going with your answers as, after twenty correct answers, you will get a tasty reward.

Always remember to use your CAPS lock (press CTRL and T together) to give capital letters. You may be asked questions alternately from French to English then English to French.

## Programming Hints

Some of the data we have used is very simple so you can change lines 600 onwards to insert your own, or have someone else program, harder examples.

## Program

```

10 REM ENGLISH/FRENCH
20 CLS
40 PAPER 7:INK 0
70 DIM E$(20)
80 DIM F$(20)
100 FOR J=1 TO 20
110 READ E$(J)
120 READ F$(J)
130 NEXT J
132 PRINT:PRINT
140 INPUT "WHAT IS YOUR NAME";N$
142 IF LEN(N$)>8 THEN N$=LEFT$(N$,8)
150 WAIT 200
154 CLS
160 PRINT@5,0;CHR$(147);CHR$(128);"SCOR
E: ";SC;CHR$(151)
170 PRINT @17,0;CHR$(147);CHR$(128);"QU
ESTIONS: ";QN;CHR$(151)
180 R=INT(RND(1)*20)+1
190 IF RND(1)>0.5 THEN 330
200 PRINT @5,6;CHR$(146);CHR$(128);"WHA
T IS THE FRENCH WORD";CHR$(151)
202 PRINT @5,7;CHR$(146);CHR$(128);"FOR
";E$(R);"?"CHR$(151)
204 PRINT @2,20;"";
210 INPUT A$
220 QN=QN+1

```

```


240 IF A$=F$(R) THEN 280
250 PLAY 1,0,0,0:MUSIC 1,1,1,10:WAIT 70
:PLAY 0,0,0,0
260 PRINT @5,12;CHR$(149);CHR$(128);"NO
";N$;" THE WORD IS";CHR$(151)
262 PRINT @5,13;CHR$(149);CHR$(128);F$(
R);CHR$(151)
270 GOTO 150
280 PLAY 1,0,0,0:MUSIC 1,2,9,10:WAIT 30
:PLAY 0,0,0,0
290 PRINT @12,10;CHR$(149);CHR$(128);"C
ORRECT! ";CHR$(151)
300 SC=SC+1
310 IF SC=20 THEN 410
320 GOTO 150
330 PRINT @5,6;CHR$(146);CHR$(128);"WHA
T IS THE ENGLISH WORD";CHR$(151)
332 PRINT @5,7;CHR$(146);CHR$(128);"FOR
";F$(R);"? ";CHR$(151)
334 PRINT @2,20;"";
340 INPUT A$
350 QN=QN+1
370 IF A$=E$(R) THEN 280
380 PLAY 1,0,0,0:MUSIC 1,1,1,10:WAIT 70
:PLAY 0,0,0,0
390 PRINT @5,12;CHR$(149);CHR$(128);"NO
";N$;" THE WORD IS ";CHR$(151)
392 PRINT @5,13;CHR$(149);CHR$(128);E$(
R);CHR$(151)
400 GOTO 150
410 PAPER 0
430 CLS
440 PRINT @6,8;CHR$(150);CHR$(128);"CON
GRATULATIONS! YOU HAVE ";CHR$(144)
450 PRINT @6,9;CHR$(150);CHR$(128);"ANS
WERED 20 QUESTIONS ";CHR$(144)
460 PRINT @6,10;CHR$(150);CHR$(128);"CO
RRECTLY. HERE IS HALF A ";CHR$(144)
464 PRINT @6,11;CHR$(150);CHR$(128);"FR
ENCH ONION TO CHEW! ";CHR$(144)
466 WAIT 500
468 PAPER 0:HIRES:CURSET 120,100,1
470 FOR R=1 TO 45 STEP 3
480 CIRCLE R,1
490 NEXT R
500 GET N$

```

510 PAPER 7  
530 CLS:TEXT:PAPER 7  
540 END  
600 DATA "TABLE","LA TABLE"  
610 DATA "CHAIR","LA CHAISE"  
620 DATA "DOOR","LA PORTE"  
630 DATA "HOUSE","LA MAISON"  
640 DATA "DOG","LE CHIEN"  
650 DATA "CAT","LE CHAT"  
660 DATA "GARDEN","LE JARDIN"  
670 DATA "COAT","LE MANTEAU"  
680 DATA "HAT","LE CHAPEAU"  
690 DATA "BICYCLE","LA BICYCLETTE"  
700 DATA "TRAIN","LE TRAIN"  
710 DATA "STATION","LA GARE"  
720 DATA "BREAD","LE PAIN"  
730 DATA "MILK","LE LAIT"  
740 DATA "CUP","LA TASSE"  
750 DATA "APPLE","LA POMME"  
760 DATA "ROAD","LA RUE"  
770 DATA "MAP","LA CARTE"  
780 DATA "SEA","LA MER"  
790 DATA "BOAT","LE BATEAU"

# 9

## Fizz/Buzz



You might have played this at school with everyone in your class making a fool of themselves but now it is up to you alone to get it right or look silly.

Every time a number can be divided by 3 you must press FIZZ and every time it can be divided by 5 you press BUZZ. Remember to press ENTER after each entry.

Example: 1, 2, FIZZ, 4, BUZZ, and so on.

### How to play

You type in the numbers and the FIZZES and BUZZES.

If you make a mistake the computer will tell you what you have done wrong and will send you back to the beginning.

Press CTRL and T to give you capital letters before you start.

At the end your computer will thank you for the game.

## Programming Hints

You can easily change the program for numbers other than 3 and 5 if you find you are getting too good for the computer. Why not try 5 and 7? The lines to change are 110, 120, 130, 160, 190, 200 and 210.

## Program

```

10 REM FIZZ BUZZ
20 CLS
30 PAPER 7: INK 0
40 C=1
70 PRINT @10,0;CHR$(147);CHR$(128);"HIGH
H SCORE: ";HS;CHR$(151)
90 PRINT:PRINT
100 INPUT A$
102 IF A$="" THEN 100
110 IF A$="FIZZ" AND C/3=INT(C/3) AND C
/5<>INT(C/5) THEN 300
120 IF A$="BUZZ" AND C/5=INT(C/5) AND C
/3<>INT(C/3) THEN 300
130 IF A$="FIZZ BUZZ" AND C/3=INT(C/3)
AND C/5=INT(C/5) THEN 300
140 IF LEFT$(A$,1)>"9" OR LEFT$(A$,1)<"
0" THEN 170
150 ANS=VAL(A$)
160 IF ANS=C AND C/3<>INT(C/3) AND C/5<
>INT(C/5) THEN 300
170 EXPLODE
180 PRINT "THAT SHOULD HAVE BEEN ";
190 IF C/3<>INT(C/3) AND C/5<>INT(C/5)
THEN PRINT C
200 IF C/3=INT(C/3) THEN PRINT "FIZZ"
210 IF C/5=INT(C/5) THEN PRINT "BUZZ"

```

```
220 PRINT:PRINT
230 PRINT "YOUR SCORE IS ";C-1
234 PRINT:PRINT
240 IF C-1>HS THEN HS=C-1:PRINT "A NEW
HIGH SCORE!":PRINT:PRINT
250 INPUT "PLAY AGAIN";Q$
260 CLS
270 IF LEFT$(Q$,1)="Y" THEN 280
272 PRINT @6,8;CHR$(149);CHR$(128);"THA
NKS FOR THE GAME!";CHR$(151):END
280 GOTO 40
300 ZAP
320 PRINT:PRINT
330 C=C+1
340 GOTO 100
```

# 10

## Highway Code

QUESTIONS: 10 SCORE: 9

On a road without a footpath,  
you should walk:

1. On either side.
2. On the left hand side.
3. On the right-hand side.

PRESS 1, 2 OR 3. JOHN

Sorry, we don't have a green giant to help you across the road or help you solve any of the questions in this game. This program is designed to make sure that you know your code before you begin to take your cycle out on the open road. It will also help you to keep your parents out of trouble as you can tell them where they shouldn't park their cars.

### How to play

The first thing your jolly green computer will ask you is your name which you should type in then press RETURN.

You will then be given a series of questions with three answers to each. You must choose the correct one if you want to get points and, in real life, stay alive on the road.



Example: The best place to cross the road is

1. a) at a bus stop
2. b) between cars
3. c) on a zebra crossing

You must press 1, 2 or 3. The quicker your response the higher the score.

The game will end after twenty questions which are at random so you may get the same one twice.

### Programming Hints

To change the selection of questions you alter the input data lines from 600 onwards if you wish to make things easier or more difficult for the player.

### Program

```

10 REM HIGHWAY CODE
40 CLS
50 PRINT:PRINT
70 DIM Q$(13)
80 DIM A$(13,3)
90 PAPER 0:INK 4
100 INPUT "WHAT IS YOUR NAME";N$
110 FOR J=1 TO 13
120 READ Q$(J)
130 FOR K=1 TO 3
140 READ A$(J,K)
150 NEXT K
160 NEXT J
170 CLS
180 PRINT @3,1;CHR$(147);CHR$(128);"QUE
STIONS: ";QN;CHR$(144)
190 PRINT @22,1;CHR$(147);CHR$(128);"SC
ORE: ";SC;CHR$(144)
200 R=INT(RND(1)*13)+1
210 IF R=RS THEN 200

```

```

220 RS=R
230 PRINT @2,7;Q$(R)
240 RA=INT(RND(1)*3)+1
250 RB=INT(RND(1)*3)+1
260 IF RB=RA THEN 250
270 RC=INT(RND(1)*3)+1
280 IF RC=RA OR RC=RB THEN 270
290 PRINT @2,10;"1. ";A$(R,RA)
300 PRINT @2,12;"2. ";A$(R,RB)
310 PRINT @2,14;"3. ";A$(R,RC)
320 PRINT @2,18;CHR$(149);CHR$(128);"PR
ESS 1, 2 OR 3, ";N$;CHR$(144)
330 QN=QN+1
340 FOR T=1 TO 300
350 R$=KEY$
360 IF T/50<>INT(T/50) THEN 370
362 PLAY 1,0,0,0
364 MUSIC 1,3,T/50+1,8
366 WAIT 30
368 PLAY 0,0,0,0
370 IF R$="1" THEN T$=A$(R,RA):GOTO 440
380 IF R$="2" THEN T$=A$(R,RB):GOTO 440
390 IF R$="3" THEN T$=A$(R,RC):GOTO 440
400 NEXT T
410 EXPLODE
420 PRINT @2,20;CHR$(146);CHR$(128);"OU
T OF TIME!";CHR$(144)
430 GOTO 520
440 IF A$(R,1)=T$ THEN 480
450 EXPLODE
460 PRINT @2,20;CHR$(146);CHR$(128);"TH
AT IS WRONG, ";N$;CHR$(144)
470 GOTO 520
480 PING
490 PRINT @2,20;CHR$(146);CHR$(128);"CO
RRECT!";CHR$(144)
492 SI=10-INT(T/100)
500 SC=SC+SI
520 WAIT 300
522 IF QN=20 THEN 540
530 GOTO 170
540 CLS
542 PLAY 1,0,0,0:MUSIC 1,3,1,10:WAIT 30
546 MUSIC 1,3,5,10:WAIT 50
548 MUSIC 1,3,9,10:WAIT 90:PLAY 0,0,0,0

```

```

550 PRINT @4,10;CHR$(150);CHR$(128);"YOUR FINAL SCORE IS ";SC;CHR$(151)
560 PAPER 7:INK 0
580 END
600 DATA "WHAT SHOULD YOU WEAR AT NIGHT?", "SOMETHING LIGHT OR REFLECTIVE."
605 DATA "DARK CLOTHES.", "AS MANY CLOTHES AS POSSIBLE."
610 DATA "WHICH IS THE SAFEST PLACE TO CROSS THE ROAD?", "A ZEBRA CROSSING."
615 DATA "A DUAL CARRIAGEWAY.", "BETWEEN PARKED CARS."
620 DATA "WHERE SHOULD YOU STAND BEFORE CROSSING THE ROAD?"
625 DATA "ON THE PAVEMENT, A LITTLE WAY FROM THE KERB."
628 DATA "BEHIND A PARKED CAR.", "IN THE ROAD."
630 DATA "WHERE MIGHT YOU SEE ZIG-ZAG LINES?", "NEAR A ZEBRA CROSSING."
635 DATA "NEAR TRAFFIC LIGHTS", "ON A BUSY ROAD."
640 DATA "ON A ROAD WITHOUT A FOOTPATH, YOU SHOULD WALK:"
645 DATA "ON THE RIGHT HAND SIDE.", "ON THE LEFT HAND SIDE."
648 DATA "ON EITHER SIDE."
650 DATA "WHAT IS THE BEST WAY TO CROSS A CLEAR ROAD?"
655 DATA "WALK STRAIGHT ACROSS.", "RUN.", "WALK DIAGONALLY."
660 DATA "WHEN WILL A CAR TAKE LONGEST TO STOP.", "ON A WET ROAD."
665 DATA "ON A DRY ROAD.", "ON A BUSY ROAD."
670 DATA "AT A PELICAN CROSSING, YOU SEE A RED MAN, WHAT DO YOU DO?"
675 DATA "WAIT.", "RUN QUICKLY ACROSS.", "WALK ACROSS."
680 DATA "WHERE SHOULD YOU GET ON A BUS?", "AT A BUS STOP."
685 DATA "AT TRAFFIC LIGHTS.", "ANYWHERE IT HAS STOPPED."
690 DATA "WHAT MUST YOU HAVE ON A CYCLE AT NIGHT?"

```

695 DATA "FRONT AND REAR LAMPS AND A REFLECTOR.", "FRONT AND REAR LAMPS."

698 DATA "A FRONT LAMP AND A REFLECTOR."  
"

700 DATA "WHEN YOU ARE WITH A DOG, SHOULD IT BE:", "ON A LEAD."

705 DATA "IN THE ROAD.", "RUNNING FREE."

710 DATA "RED AND AMBER LIGHTS AT TRAFFIC LIGHTS MEANS:"

715 DATA "TRAFFIC ABOUT TO START.", "TRAFFIC STOPPING."

718 DATA "TRAFFIC MOVING BOTH WAYS."


720 DATA "ON A CYCLE AT TRAFFIC LIGHTS SHOWING RED WHAT SHOULD YOU DO?"

725 DATA "WAIT UNTIL THEY ARE GREEN.", "CYCLE THROUGH."

728 DATA "GET OFF AND WALK THROUGH."

# 11

## Guess the Number



You have the chance here to see if you can guess the number your computer has chosen for you.

The game will begin with a number between 1 and 100 but as you become an expert at guessing the number your computer will start to widen the range to between 1 and 120, and then 140 and so on.

### How to play

Your friendly computer will ask you to guess the number it has thought of and you enter your first guess. Your computer will then tell you whether you are too high or too low with your try and will recommend your next move to you.

If your answer is quick and correct then the game becomes harder

with a wider range of numbers but if you are slow then your computer will wind down to help you.

## Programming Hints

You can make the game easier or more difficult to start with if you decrease or increase the value of 'max' in line 50.

## Program

```

10 REM GUESS THE NUMBER
20 CLS
30 PAPER 7:INK 0
50 MAX=100
80 PRINT @11,2;"HELLO!"
84 WAIT 150
90 PRINT @2,6;"I AM GOING TO THINK OF A
NUMBER "
91 PRINT @2,7;"BETWEEN 1 AND";MAX;"."
92 PRINT @2,8;"YOU TRY AND GUESS WHAT I
T IS."
94 WAIT 150
100 PRINT@9,21;CHR$(149);"THINKING...";
CHR$(151)
110 R=INT(RND(1)*MAX)+1
120 WAIT 500
124 PRINT @9,21;" "
130 PRINT @2,10;"O.K. WHAT NUMBER AM I
THINKING OF?"
140 PRINT:PRINT
150 INPUT ANS
154 PRINT:PRINT
160 TR=TR+1
170 IF ANS<=R THEN 180
174 PRINT "YOUR GUESS IS TOO HIGH. TRY
A LOWER NUMBER.":GOTO 140
180 IF ANS=R THEN 190
184 PRINT "YOUR GUESS IS TOO LOW. TRY A
HIGHER NUMBER.":GOTO 140
190 PLAY1,0,0,0:MUSIC 1,3,9,10:WAIT 30
200 MUSIC 1,4,6,10:WAIT 50:PLAY 0,0,0,0
210 PRINT "THAT IS THE NUMBER!"

```

52 *Guess the Number*

```
214 PRINT:PRINT
216 WAIT 400
218 TP=MAX
220 C=C+1
230 TP=TP/2
240 IF TP>1 THEN 220
250 PRINT "YOU GUESSED IN ";TR;" TRIES.
"
```

```
252 PRINT:PRINT
260 IF TR<C+2 THEN 310
270 MAX=MAX-20
280 IF MAX<20 THEN MAX=20:GOTO 330
290 PRINT "THAT WAS A BIT SLOW! I WILL
REDUCE THE MAXIMUM NUMBER TO ";MAX
300 GOTO 330
310 MAX=MAX+20
320 PRINT "THAT IS VERY GOOD! I WILL IN
CREASE THE MAXIMUM NUMBER TO ";MAX
330 C=0
340 TR=0
350 WAIT 400
360 CLS
370 GOTO 90
```

# 12

## Word Mix

WORDS: 2      SCORE: 4

BMLAOCBNINGK

WHAT ARE THE TWO WORDS?

THE TWO WORDS WERE  
BLACK              MORNING

If you ever get words mixed together when you are thinking about something else and have to go back and rewrite your essay then this game will be easy for you.

We have mixed one word through another but we haven't jumbled them up - if you see what we mean.

Here is an example: ABFOOUUTR

spells ABOUT FOUR

As you can see both words are mixed together but not jumbled. You have to have a sharp eye to pick them out and separate them.



**How to play**

The computer will put up on your screen two words mixed as in the example above.

Type in the first word and press RETURN then the second word and RETURN.

You can use lower case - capitals aren't necessary.

You will receive one point for each correct word and the score will appear on the top of your screen.

If you cannot solve the riddle type QUIT to give up.

**Programming Hints**

If you wish to make the selection of words even more difficult then change the data in lines 450 onward. If you increase the number of words make sure that you change the 50 in lines 60, 80, 140, 150 to match the total number of words, e.g. if you had 100 words, line 60 would be:

```
60 DIM W$(100)
```

**Program**

```
10 REM WORD MIX
30 PAPER 7: INK 0
60 DIM W$(50)
80 FOR J=1 TO 50
90 READ W$(J)
100 NEXT J
110 CLS
120 PRINT @5,1;CHR$(147);"WORDS: ";TR;C
HR$(151)
130 PRINT @19,1;CHR$(147);"SCORE: ";SC;
CHR$(151)
140 R1=INT(RND(1)*50)+1
150 R2=INT(RND(1)*50)+1
160 IF R2=R1 THEN 150
170 M1=1
```

```

180 M2=1
190 PRINT @12,9;
200 IF RND(1)>0.5 THEN 246
208 IF M1>10 THEN 280
210 IF MID$(W$(R1),M1,1)=" " THEN 230
220 PRINT MID$(W$(R1),M1,1);
230 M1=M1+1
240 GOTO 280
246 IF M2>10 THEN 280
250 IF MID$(W$(R2),M2,1)=" " THEN 270
260 PRINT MID$(W$(R2),M2,1);
270 M2=M2+1
280 IF M1<10 OR M2<10 THEN 200
290 PRINT @6,15;CHR$(149);"WHAT ARE THE
TWO WORDS?";CHR$(151)
300 PRINT @2,20;"":INPUT A$,B$
304 PRINT @2,20;"
"
306 PRINT@4,15;"
"
310 TR=TR+1
314 IF A$="QUIT" THEN 420
340 IF A$<>W$(R1) AND A$<>W$(R2) THEN 3
70
350 PRINT @3,14;CHR$(146);"YOUR FIRST W
ORD IS CORRECT!";CHR$(151):PING
360 A$=""
370 IF B$<>W$(R1) AND B$<>W$(R2) THEN 4
00
380 PRINT @3,16;CHR$(146);"YOUR SECOND
WORD IS CORRECT!";CHR$(151):PING
390 B$=""
394 WAIT 200
400 SC=SC-(A$="")-(B$="")
410 IF A$="" AND B$="" THEN 110
420 PRINT @8,18;CHR$(149);"THE TWO WORD
S WERE";CHR$(151)
430 PRINT @8,20;CHR$(149);W$(R1);CHR$(1
51)
432 PRINT @19,20;CHR$(149);W$(R2);CHR$(
151)
434 WAIT 500
440 GOTO 110
450 DATA "BANK","HOUR","MOST"


```

460 DATA "YARD", "FOUR", "SETTLE"  
 470 DATA "MORNING", "TIME", "FOUND"  
 480 DATA "HOOF", "SAD", "BLACK"  
 490 DATA "ANIMAL", "ABOUT", "SOLDIER"  
 500 DATA "COLONEL", "RUMOUR", "ENGINE"  
 510 DATA "LEG", "CASTLE", "DIGEST"  
 520 DATA "GUITAR", "DOG", "BUTTER"  
 530 DATA "BRICK", "SPACE", "DUCK"  
 540 DATA "AEROPLANE", "BATTERY", "SPARROW

550 DATA "PRINT", "STAR", "SYSTEM"  
 560 DATA "CANARY", "BASE", "WHITE"  
 570 DATA "TABLE", "CHAIR", "MAGIC"  
 580 DATA "LAND", "JUSTIFY", "HABIT"  
 590 DATA "MECHANIC", "CYLINDER", "OIL"  
 600 DATA "SPADE", "MARKET", "TRACTOR"  
 610 DATA "ROBUST", "VARIETY"

# 13

## Constellations



This is a great game for learning how to make money off your pals by asking them at night if they know which star group is which in the sky.

It is also very handy to know your stars as you never know when you will need to navigate your way out of a crocodile-infested swamp - or drive to London in the dark. The star at the end of the tail of Ursa Minor, or the little bear, is called the pole star and will always be to your North.

### How to play

The computer will show you the shape of some of the main star groups to be found in the skies around us and will give you the Latin and the common names for each group.

## 58 Constellations

The screen will then show the stars without any names and will ask you to type in your answer in CAPS.

If your guess is wrong the screen will light up and show you both the names. Press CTRL and C together to stop the game.

Happy stargazing.


### Program

```
10 REM CONSTELLATIONS
20 PAPER 0:INK 6
30 HIRES
50 A=70:B=90
54 TT=0
60 FOR C=1 TO 7
62 IF TT=0 THEN 70
64 R=INT(RND(1)*7)+1
66 GOSUB 300
70 READ X,Y
80 IF X=-99 THEN 140
100 CURSET A+X,B+Y,1
110 GOTO 70
140 READ M$,N$
142 IF TT=0 THEN 150
144 INPUT "WHICH IS THIS";A$
146 IF A$=M$ OR A$=N$ THEN 400
148 EXPLODE
150 PRINT M$
160 PRINT N$
170 WAIT 1000
172 HIRES
180 NEXT C
190 TT=1
192 GOTO 60
200 DATA 0,0,24,-12,39,-9,63,-10,75,0,1
05,-12,97,-33,-99,0
205 DATA "URSA MAJOR","GREAT BEAR"
210 DATA 0,0,12,1,3,-19,13,-18,24,-33,4
0,-42,57,-48,-99,0
215 DATA "URSA MINOR","LITTLE BEAR"
220 DATA 0,0,6,18,18,18,24,37,44,28,-99
,0,"CASSIOPEIA"," "
```

```
230 DATA 0,1,5,-4,-5,5,-6,35,12,-30,26,  
30,-20,-33,-99,0  
235 DATA "ORION","THE HUNTER"  
240 DATA 0,0,7,2,14,1,20,-2,25,-10,20,-  
19,-99,0  
245 DATA "CORONA BOREALIS","NORTHERN CR  
OWN"  
250 DATA 0,0,21,9,39,12,38,19,-99,0,"SA  
GITTA","THE ARROW"  
260 DATA 0,0,25,-12,34,-55,47,-35,90,9,  
72,-8,66,-15,74,-52,-99,0  
265 DATA "CYGNUS","THE SWAN"  
300 RESTORE  
310 FOR J=1 TO R-1  
320 READ NU  
330 IF NU=-99 THEN 350  
340 GOTO 320  
350 READ NU,Z$,Z$  
360 NEXT J  
370 RETURN  
400 PING  
410 PRINT "THAT IS CORRECT!"  
420 WAIT 400  
430 GOTO 172
```

# 14

## History Quiz



Here is a chance to catch up on your history lessons without looking stupid because you have got the answer wrong. If you are working with your own computer no one will know that you need more practice with your history.

This is very like the science quiz which offers multiple answers for which you must make the correct choice. You can also alter the questions in this program as explained at the end.

### How to play

The computer will ask you for your name and you type it in and press RETURN.

A choice of statements will appear on the screen and you will be asked to choose the correct answer.

Example:

The Napoleonic wars were fought between?

1. The French and English
2. The English and Americans
3. The French and Italians

To answer press 1, 2 or 3

If you are correct the computer will tell you and if you are wrong the computer will print 'That is wrong'.

### Programming Hints

The lines to change for your own questions are from 600 onwards. If you increase the number of questions you will need to change the 20 in lines 70, 80, 110 and 200 to match the total number of questions. To give yourself more or less time for each question, increase or decrease the number in line 340 in multiples of 100.

### Program

```

10 REM HISTORY QUIZ
40 CLS
50 PRINT:PRINT
70 DIM Q$(20)
80 DIM A$(20,3)
90 PAPER 0:INK 4
100 INPUT "WHAT IS YOUR NAME";N$
110 FOR J=1 TO 20
120 READ Q$(J)
130 FOR K=1 TO 3
140 READ A$(J,K)
150 NEXT K
160 NEXT J

```



```

170 CLS
180 PRINT @3,1;CHR$(147);CHR$(128);"QUE
STIONS: ";QN;CHR$(144)
190 PRINT @22,1;CHR$(147);CHR$(128);"SC
ORE: ";SC;CHR$(144)
200 R=INT(RND(1)*20)+1
210 IF R=RS THEN 200
220 RS=R
230 PRINT @2,7;Q$(R)
240 RA=INT(RND(1)*3)+1
250 RB=INT(RND(1)*3)+1
260 IF RB=RA THEN 250
270 RC=INT(RND(1)*3)+1
280 IF RC=RA OR RC=RB THEN 270
290 PRINT @2,10;"1. ";A$(R,RA)
300 PRINT @2,12;"2. ";A$(R,RB)
310 PRINT @2,14;"3. ";A$(R,RC)
320 PRINT @2,18;CHR$(149);CHR$(128);"PR
ESS 1, 2 OR 3, ";N$;CHR$(144)
330 QN=QN+1
340 FOR T=1 TO 300
350 R$=KEY$
360 IF T/50<>INT(T/50) THEN 370
362 PLAY 1,0,0,0
364 MUSIC 1,3,T/50+1,8
366 WAIT 30
368 PLAY 0,0,0,0
370 IF R$="1" THEN T$=A$(R,RA):GOTO 440
380 IF R$="2" THEN T$=A$(R,RB):GOTO 440
390 IF R$="3" THEN T$=A$(R,RC):GOTO 440
400 NEXT T
410 EXPLODE
420 PRINT @2,20;CHR$(146);CHR$(128);"OU
T OF TIME!";CHR$(144)
430 GOTO 520
440 IF A$(R,1)=T$ THEN 480
450 EXPLODE
460 PRINT @2,20;CHR$(146);CHR$(128);"TH
AT IS WRONG, ";N$;CHR$(144)
470 GOTO 520
480 PING
490 PRINT @2,20;CHR$(146);CHR$(128);"CO
RRECT!";CHR$(144)
492 SI=10-INT(T/35)
500 SC=SC+SI
520 WAIT 300

```

```

522 IF QN=20 THEN 540
530 GOTO 170
540 CLS
542 PLAY 1,0,0,0:MUSIC 1,3,1,10:WAIT 30
546 MUSIC 1,3,5,10:WAIT 50
548 MUSIC 1,3,9,10:WAIT 90:PLAY 0,0,0,0

550 PRINT @4,10;CHR$(150);CHR$(128);"YO
UR FINAL SCORE IS ";SC;CHR$(151)
560 PAPER 7:INK 0
580 END
600 DATA "WHEN DID JULIUS CAESAR FIRST
COME TO BRITAIN?","55 BC","1914"
605 DATA "1066"
610 DATA "WHO REIGNED BEFORE QUEEN ELIZ
ABETH II","GEORGE VI"
615 DATA "QUEEN VICTORIA","EDWARD VII"
620 DATA "WHO INTRODUCED THE PENNY POST
?","ROWLAND HILL","JULIUS CAESAR"
625 DATA "WALTER RALEIGH"
630 DATA "WHERE DID THE GREAT FIRE OF L
ONDON START?","PUDDING LANE"
635 DATA "OXFORD CIRCUS","FURNACE LANE"
640 DATA "WHEN WAS THE BATTLE OF HASTIN
GS?","1066","1943","1841"
650 DATA "WHY WRE THE EGYPTIAN PYRAMIDS
BUILT?","AS TOMBS","AS HOUSES"
655 DATA "FOR GRAIN STORAGE"
660 DATA "WHO WAS SENT TO AUSTRALIA IN
THE LATE EIGHTEENTH CENTURY?"
665 DATA "CONVICTS","PRISONERS OF WAR",
"SLAVES"
670 DATA "WHICH CENTURY ARE WE IN NOW?"
,"TWENTIETH","NINETEENTH"
675 DATA "EIGHTEENTH"
680 DATA "WHO WAS DISRAELI?","A PRIME M
INISTER","A KING","A COMPOSER"
690 DATA "WHO INVENTED THE TELEPHONE?",
"BELL","COOK","BAIRD"
700 DATA "WHO WAS THE FIRST MAN IN SPAC
E?","YURI GAGARIN","NEIL ARMSTRONG"
705 DATA "LEONARDO DA VINCI"
710 DATA "WHEN DID THE FIRST WORLD WAR
START?","1914","1939","1918"
720 DATA "WHO WAS SIR ROBERT WALPOLE?",
"THE FIRST PRIME MINISTER"

```

725 DATA "INVENTOR OF THE LOCOMOTIVE","  
DISCOVERER OF AMERICA"

730 DATA "WITH WHICH ANIMALS DID HANNIB  
AL CROSS THE ALPS?","ELEPHANTS"

735 DATA "CAMELS","LLAMAS"

740 DATA "WHO STARTED THE POLICE FORCE?  
","PEEL","NELSON","GLADSTONE"

750 DATA "WHO DEFEATED THE SPANISH ARMA  
DA?","FRANCIS DRAKE"

755 DATA "WALTER RALEIGH","HENRY VIII"

760 DATA "WHAT WAS CHAIN-MAIL?","ARMOUR  
","AN EARLY POSTAL SERVICE"

765 DATA "A SHIPS ANCHOR CHAIN"

770 DATA "HOW MANY WIVES DID HENRY VIII  
HAVE?","SIX","EIGHT","TWO"


780 DATA "WHAT ANIMAL SPREAD THE PLAGUE  
OF 1665?","RAT","MOUSE","HORSE"

790 DATA "WHO WAS THE FIRST PERSON TO W  
ALK ON THE MOON?","ARMSTRONG"

795 DATA "GAGARIN","SHEPHERD"

# 15

## Spelling Test



Well, you must have expected to find a spelling test somewhere in the book and here it is.

### How to play

Your computer will put a word on the screen for a few seconds and then blank the screen and ask you to spell the same word correctly.

You must use upper case (capital letters) for this game. Remember you can change from small letters to capitals and vice versa by pressing CTRL and T together.

**Programming Hints**

If you think the time that you see the word on the screen is too long or too short you can alter the pause in line 170. If you feel that the words are too easy for a bright young thing like you then ask your parents to type in some more difficult words from line 400 onward. If you add more words, change the 50 in lines 70, 80, and 120 to match the total number of words.

The computer will ask its questions at random so you shouldn't really know what is coming up next.

**Program**

```

10 REM SPELLING TEST
40 PAPER 7:INK 0
70 DIM W$(50)
80 FOR J=1 TO 50
90 READ W$(J)
100 NEXT J
110 FOR N=1 TO 20
120 R=INT(RND(1)*50)+1
130 CLS
140 PRINT @4,0;CHR$(147);"SCORE: ";SC;C
HR$(151)
150 PRINT @18,0;CHR$(147);"HIGH SCORE:
";HS;CHR$(151)
160 PRINT @12,8;W$(R)
170 WAIT 140
180 PRINT @12,8;" "
190 PRINT @6,16;CHR$(150);"WHAT WAS THE
WORD?";CHR$(151)
194 PRINT:PRINT
200 INPUT A$
210 IF A$=W$(R) THEN 270
220 EXPLODE
230 PRINT @8,8;CHR$(150);"THE WORD WAS
";W$(R);CHR$(151)
244 PRINT @3,16;"
"
250 WAIT 700
260 GOTO 300
270 PING

```


```

280 SC=SC+10
300 NEXT N
310 IF SC>HS THEN HS=SC
312 CLS
314 PRINT @3,8;"YOUR FINAL SCORE WAS ";
SC
316 IF SC=HS THEN PRINT @2,10;"THIS IS
THE HIGHEST SCORE TODAY."
318 PRINT @ 2,18;"";
320 INPUT "PLAY AGAIN";Q$
330 IF LEFT$(Q$,1)="N" THEN 380
340 SC=0
360 GOTO 110
380 END
400 DATA "QUIZ","MESSAGE","PAVEMENT"
410 DATA "BICYCLE","SPECIAL","BENEATH"
420 DATA "MOUNTAIN","LISTEN","SCHOOL"
430 DATA "TOMORROW","BUSINESS","ADDRESS
"
440 DATA "PARALLEL","HEIGHT","LENGTH"
450 DATA "CEILING","EXPERT","KETTLE"
460 DATA "COLONEL","SURPRISE","FORECAST
"
470 DATA "ATTACH","RHUBARB","MERINGUE"
480 DATA "DAFFODIL","KNOWLEDGE","YACHT"
490 DATA "TONGUE","MINIATURE","DINGHY"
500 DATA "AMATEUR","PUNCTUAL","ILLOGICA
L"
510 DATA "GIRAFFE","PARSLEY","TRIANGLE"
520 DATA "LEGIBLE","MOSAIC","DISCIPLE"
530 DATA "AMEND","GUITAR","BELIEVE"
540 DATA "STATION","PRESENCE","SAVIOUR"
550 DATA "ALCOHOL","CABARET","SYLLABLE"
560 DATA "AQUATIC","PNEUMATIC"

```

# 16

## Speed Reading



This is a game for those show-offs amongst you who can remember everything, write it down and get it right every time.

A sentence will appear on the screen for a few seconds and you must remember the sentence and the correct spelling of the words if you are going to get your points.

There is one big snag for clever little people though - the better your answers the faster the computer goes and you will only get to see the sentence for a split second before you have to answer.

### How to play

Your computer will put the line up on the screen and ask you to type it in after a short pause.

If you get it wrong the screen will tell you.

## Programming Hints

If you want to change the sentences to make it more difficult for yourself, or your friends, the lines to change are from 400 onward.

## Program

```

10 REM SPEED READING
30 PAPER 7: INK 0
50 TM=250
70 DIM W$(30)
80 FOR J=1 TO 30
90 READ W$(J)
100 NEXT J
110 FOR N=1 TO 20
120 R=INT(RND(1)*30)+1
130 CLS
140 PRINT @4,0;CHR$(147);"SCORE: ";SC;C
HR$(151)
150 PRINT @18,0;CHR$(147);"HIGH SCORE:
";HS;CHR$(151)
160 PRINT @2,8;W$(R)
170 WAIT TM-SC
180 PRINT @2,8;"
"
190 PRINT @6,16;CHR$(150);"WHAT WAS THE
SENTENCE?";CHR$(151)
194 PRINT:PRINT
200 INPUT A$
210 IF A$=W$(R) THEN 270
220 EXPLODE
230 PRINT @9,8;CHR$(150);"THE SENTENCE
WAS ";CHR$(151):PRINT
240 PRINT CHR$(150);W$(R);CHR$(151)
244 PRINT @3,16;"
"
250 WAIT 700
260 GOTO 300
270 PING
280 SC=SC+10
300 NEXT N


```



```
310 IF SC>HS THEN HS=SC
312 CLS
314 PRINT @3,8;"YOUR FINAL SCORE WAS ";
SC
316 IF SC=HS THEN PRINT @2,10;"THIS IS
THE HIGHEST SCORE TODAY."
318 PRINT @ 2,18;"";
320 INPUT "PLAY AGAIN";Q$
330 IF LEFT$(Q$,1)="N" THEN 380
340 SC=0
350 TM=250
360 GOTO 110
380 END
400 DATA "FOUR GREEN TURTLES"
410 DATA "LONG GREEN PENCIL"
420 DATA "THE WHEEL IS ROUND"
430 DATA "THE BOOK IS NOT HEAVY"
440 DATA "THE LEAF IS GREEN"
450 DATA "LOOK AT THE TREE"
460 DATA "PICK UP THE RABBIT"
470 DATA "EXTEND THE LADDER"
480 DATA "THE KITTEN IS UP A TREE"
490 DATA "WE WAITED A LONG TIME"
500 DATA "WE DASHED THROUGH THE GATE"
510 DATA "THE UMBRELLA IS OPEN"
520 DATA "THE UMBRELLA IS CLOSED"
530 DATA "THE BOOK IS TOO HEAVY"
540 DATA "THE SHOP IS CLOSED"
550 DATA "OPEN THE CUPBOARD"
560 DATA "RAPID READING RESEARCH"
570 DATA "ELEPHANTS ARE LARGE MAMMALS"
580 DATA "THE GRASS IS VERY GREEN"
590 DATA "THE RIVER IS FAST AND DEEP"
600 DATA "GREY COMPUTER PERIPHERAL"
610 DATA "COMPLETELY CORRECTED"
620 DATA "RESEARCH AND DEVELOPMENT"
630 DATA "UNTANGLE THE KNOT"
640 DATA "FIERY RED FIRE ENGINE"
650 DATA "CROSS-EXAMINE THE WITNESS"
660 DATA "HOLD UP THE MIRROR"
670 DATA "A LONG RUN OF LUCK"
680 DATA "THE SPANIARD OWNS A DOG"
690 DATA "SMALL PIECES OF BUTTER"
```

# 17

## Area Estimation



This game might also be called 'little boxes' as you are going to have to decide how many small boxes are required to fill the area of a large rectangular box.

### How to play

Type in the names of the 2 players, pressing RETURN after each name.

The computer will then show you a large rectangular box with one small box set in a corner.

You must now decide how many small boxes you require to fill the larger box.

## 72 *Area Estimation*

Type in your estimated number and press RETURN.

If you are wrong the computer will fill in one line on the upright (perpendicular) side of the box and ask you to estimate again on the totals required.

If you still don't estimate correctly then the bottom line (horizontal) will be completed.

You should now be able to estimate the total by multiplying the two columns together.

Should you get it right this time you'll get a 'happy' bleep from your computer and you will move on to the next shape. A wrong answer at this stage will mean that the computer will fill in all the boxes and ask you again for the total. The computer will not let you move on until you have provided the correct results. The first player to reach a score of 10 wins the game.

### **Programming Hints**

You can make the game a little easier by reducing the numbers in lines 90 and 110.

### **Program**

```
10 REM AREA ESTIMATION
32 INK 7:PAPER 0
34 CLS
40 PRINT @2,10;"WHAT ARE THE PLAYERS NAMES?"
50 INPUT "PLAYER ONE:";P$(1)
60 INPUT "PLAYER TWO:";P$(2)
90 HT=INT(RND(1)*10)+3
100 H=HT*10
110 WT=INT(RND(1)*12)+3
120 W=WT*10
122 HIRES
130 CURSET 15,155,1
140 DRAW 0,-H,1
150 DRAW W,0,1
```

```

160 DRAW 0,H,1
170 DRAW -W,0,1
180 GOSUB 600
190 FOR N=1 TO 2
200 PRINT P$(N); " HOW MANY SMALL SQUARE
S IN"
204 PRINT "THE LARGE RECTANGLE";
210 INPUT ANS
214 PRINT
220 IF ANS=HT*WT THEN 260
230 EXPLODE
240 WA=WA+1
250 GOTO 320
260 PING
270 S(N)=S(N)+4-WA
280 PRINT P$(1); " "; S(1); "      "; P$(2); "
"; S(2)
282 IF S(1)>9 OR S(2)>9 THEN 700
290 WA=0
294 WAIT 500
300 GOTO 90
320 IF WA>1 THEN 380
330 FOR J=20 TO H+15 STEP 10
340 CURSET 15,175-J,1
350 GOSUB 600
360 NEXT J
370 GOTO 500
380 IF WA>2 THEN 430
390 FOR J=15 TO W+10 STEP 10
400 CURSET J,155,1
410 GOSUB 600
420 NEXT J
422 GOTO 500
430 IF WA>3 THEN 500
440 FOR J=20 TO H+15 STEP 10
450 FOR K=15 TO W+10 STEP 10
460 CURSET K,175-J,1
470 GOSUB 600
480 NEXT K
490 NEXT J
500 NEXT N
510 GOTO 190
600 DRAW 0,-10,1
610 DRAW 10,0,1
620 DRAW 0,10,1


```

#### 74 *Area Estimation*

```
630 DRAW -10,0,1
640 RETURN
700 TEXT
750 IF S(2)>9 THEN 780
760 PRINT @7,10;CHR$(129);CHR$(27);"L
";P$(1);" WINS!"
770 GOTO 800
780 PRINT @7,10;CHR$(129);CHR$(27);"L
";P$(2);" WINS!"
800 END
```

# 18

## Number Series



This is a game where you have to think ahead and predict the next numbers to appear on the screen.

Your computer will start to print a string of numbers such as 11, 22, 33, 44 and will then stop and ask you to predict the next numbers which are, of course, 55.

### How to play

As soon as you think you know the next number, press any key to stop the sequence then type in your answer and press RETURN.

The correct answer will be rewarded by scoring a point on the board and a wrong answer will mean that the computer will show you the right number before moving on to the next string of

## 76 *Number Series*

numbers. Remember that the quicker you stop the sequence, the more points you score. As soon as you reach 50 points, the computer will stop and give you your average score.

### Programming Notes

As the program is set to choose numbers at random you may sometimes get the same series twice running which is lucky for you as you should know the right answer.

### Program


```
10 REM NUMBER SERIES
30 PAPER 0: INK 7
52 CLS
60 PRINT @4,1;CHR$(131);"TRIES: ";TR;"
    SCORE: ";SC
61 IF SC>=50 THEN 300
62 ST=1
70 R=INT(RND(1)*30)-15
80 IF R=0 THEN 70
90 IF R<0 THEN ST=100
100 PRINT @4,10;CHR$(132);ST;" ";
110 FOR J=1 TO 5
112 WAIT 200
114 A$=KEY$
116 IF A$<>" " THEN 144
120 ST=ST+R
130 PRINT CHR$(128+J);ST;
132 PLAY 1,0,0,0:MUSIC 1,2,J*2,10:WAIT
20:PLAY 0,0,0,0
140 NEXT J
144 PRINT:PRINT @5,18;CHR$(130);"WHAT I
S THE NEXT NUMBER";
150 INPUT ANS
160 TR=TR+1
170 IF ANS=ST+R THEN 220
180 PLAY 1,0,0,0:MUSIC 1,1,5,10:WAIT 10
0:PLAY 0,0,0,0
190 PRINT @7,15;CHR$(133);"THE NEXT NUM
BER IS ";ST+R
200 WAIT 300
```

```
210 GOTO 52
220 PLAY 1,0,0,0:MUSIC 1,3,9,10:WAIT 30
:PLAY 0,0,0,0
230 PRINT @14,15;CHR$(129);"CORRECT!"
240 SC=SC+7-J
250 WAIT 100
260 GOTO 52
300 PLAY 1,0,0,0
304 FOR J=1 TO 12
310 MUSIC 1,2,J,10
320 WAIT 30
330 NEXT J
334 WAIT 50
340 PLAY 0,0,0,0
360 CLS
370 PRINT @7,10;"YOUR AVERAGE SCORE WAS
"
380 PRINT @8,13;SC;" / ";TR;" = ";SC/TR
```



# 19

## Morse Code



If you are planning to become a sea captain or a pilot then this is the game for you. Even if you are just planning to help your parents on a yachting weekend you will be able to show off with this one.

There is plenty of fun here as well as you can set your own speed of reply to the computer; set your own messages; ask other people to translate what you have sent.

### How to play

The computer will put up on the screen a menu which is a list of items from which you can make a choice. Menu is

1. Send a message
2. Random letters
3. Words

The computer will ask you to choose from 1, 2 or 3.

You will then be asked to choose a speed which you would like from 1 to 100. As 100 is for professional morse code operators we suggest you start at 10.

Type in 10 and press RETURN

The computer will then ask you for your message.

If you have chosen 1 you could type in MAY THE FORCE BE WITH YOU, remembering to use capitals, press ENTER and the computer will play the message back to you in morse code.

When the message is complete the computer will return to the menu and ask you for your next choice.

No. 2 will mean that the computer will put random letters on the screen whilst sounding them at the same time. If you choose No. 3, words, the computer will sound out a word, at the speed you choose, which you must identify. When the computer completes the word it will ask you to type in the answer. If you are right - well done. If you are wrong however you will hear a nasty 'raspberry' and the computer will show you the correct answer before moving to another word.

Remember that your knowledge of morse code could save lives.

## Programming Notes

The words in the program can be changed when typing in the data statements in lines 1050 to 1080.

Notice how the morse code is stored in lines 1000 to 1030. Each of the groups of numbers represents one letter of the alphabet. For example, the letter 'A' is ·—, each 1 is a dot and each 3 is a dash.

### Program

```

10 REM MORSE CODE
18 PAPER 7:INK 0
20 CLS
40 DIM M$(26)
42 DIM W$(20)
50 FOR J=1 TO 26
60 READ M$(J)
70 NEXT J
80 PRINT @11,1;CHR$(145);"MORSE CODE";CHR$(151)
90 PRINT @2,6;CHR$(147);"1. SEND MESSAGE";CHR$(151)
100 PRINT @2,8;CHR$(150);"2. RANDOM LETTERS";CHR$(151)
110 PRINT @2,10;CHR$(146);"3. WORDS";CHR$(151)
120 PRINT @2,14;CHR$(149);"PRESS 1, 2 OR 3";CHR$(151)
124 PRINT:PRINT
130 A$=KEY$
132 IF A$="" THEN 130
136 PING
140 IF A$<"1" OR A$>"3" THEN 130
142 INPUT "SPEED (1 TO 100)";SP
143 IF SP>100 OR SP<1 THEN 142
145 CLS
148 PRINT @12,1;CHR$(145);"SPEED";SP;CHR$(151)
150 IF A$<>"1" THEN 170
160 GOTO 400
170 IF A$<>"2" THEN 190
180 GOTO 590
190 FOR J=1 TO 20
200 READ W$(J)
210 NEXT J
215 FOR N=1 TO 10
220 R=INT(RND(1)*20)+1
230 FOR J=1 TO LEN(W$(R))

```

```

240 L$=MID$(W$(R),J,1)
245 IF L$=" " THEN 300
250 L=ASC(L$)-64
260 FOR K=1 TO LEN(M$(L))
262 IF MID$(M$(L),K,1)="0" THEN 280
270 MUSIC 1,3,9,10:PLAY 1,0,0,0
274 WAIT 9*VAL(MID$(M$(L),K,1))+2:PLAY
0,0,0,0
276 WAIT 3
280 NEXT K
290 WAIT 160-SP
300 NEXT J
304 PRINT:PRINT
308 INPUT "WHAT WAS THE WORD";Q$(1)
310 IF Q$(1)<>W$(R) THEN 316
312 PING
314 PRINT @12,11;CHR$(149);"CORRECT!";C
HR$(151)
315 GOTO 318
316 EXPLODE
317 PRINT @13,11;CHR$(149);"WRONG!";CHR
$(151)
318 PRINT @10,13;CHR$(146);"THE WORD WA
S";CHR$(151)
320 PRINT @10,15;CHR$(146);W$(R);CHR$(1
51)
330 WAIT 500
334 CLS
340 NEXT N
350 GOTO 90
400 PRINT @4,15;CHR$(147);"WHAT IS YOUR
MESSAGE?";CHR$(151)
402 PRINT:PRINT
406 INPUT I$
410 FOR J=1 TO LEN(I$)
420 L$=MID$(I$,J,1)
422 IF L$=" " THEN WAIT 200:GOTO 500
424 IF L$<"A" OR L$>"Z" THEN 510
430 L=ASC(L$)-64
440 FOR K=1 TO LEN(M$(L))
442 IF MID$(M$(L),K,1)="0" THEN 470
444 MUSIC 1,3,9,10:PLAY 1,0,0,0
450 WAIT 9*VAL(MID$(M$(L),K,1))+2:PLAY
0,0,0,0
460 WAIT 3


```

```

470 NEXT K
474 PRINT @1+J,4;L$
480 WAIT 160-SP
500 NEXT J
510 WAIT 500
520 CLS
530 GOTO 90
590 PRINT @2,4;
600 FOR J=1 TO 50
610 R=INT(RND(1)*26)+1
620 PRINT CHR$(150);CHR$(R+64);CHR$(151
);
630 FOR K=1 TO LEN(M$(R))
640 IF MID$(M$(R),K,1)="0" THEN 662
644 MUSIC 1,3,9,10:PLAY 1,0,0,0
650 WAIT 9*VAL(MID$(M$(R),K,1))+2:PLAY
0,0,0,0
660 WAIT 3
662 NEXT K
670 WAIT 160-SP
680 NEXT J
690 WAIT 500
700 CLS
710 GOTO 90
1000 DATA "1300","3111","3131","3110","
1000","1131","3310"
1010 DATA "1111","1100","1333","3130","
1311","3300","3100"
1020 DATA "3330","1331","3313","1310","
1110","3000","1130"
1030 DATA "1113","1330","3113","3133","
3311"
1050 DATA "DELICATESSEN","NECESSITY","D
ELICIOUS","PREHISTORIC","GOURMET"
1060 DATA "PHILOSOPHER","PRECIOUS","DIS
SECT","AGRICULTURE","SEMAPHORE"
1070 DATA "DIFFICULT","BUSINESS","MONAS
TERY","FRAGILE","ALPHABET"
1080 DATA "DAFFODIL","KNOWLEDGE","THROU
GH","PINEAPPLE","KANGAROO"

```

# 20 Areas



This game is a bit of a brain teaser and might drive you mad before you start to get it right. To begin with you might even need a pencil and some paper but really that is a bit of a cheat.

The object is to work out the total area of a rectangle and a triangle - together. Your parents will be highly impressed when you can tell them how much carpet they need for that funny shaped room upstairs.

## How to play

You will be asked 'One or two players?'

Type in 1 or 2 and press RETURN

## 84 Areas

You will then be asked for the player's, or players', names. RETURN as before.

The computer will then show you a red rectangle with a green triangle on top. You will be given the length of three of the sides and asked to work out the total area.

If you do not get the answer correct first time the computer will then ask you to give the area of the rectangle first. If you answer this part correctly you will then be asked for the area of the triangle. If either part of your answer is wrong the computer will give you the correct total area sum.

If you get the answer right first time, clever clogs, you will be given two points but if you answer in two halves you will get only one point for your correct answers.

To help you bend the rules the formula for area is

$$C \times B + \frac{1}{2} B \times (A - C)$$

### Program

```
10 REM AREAS
20 CP=40
22 LP=90
24 PAPER 7:INK 4
30 FOR J=0 TO 15
32 READ N
34 POKE 46856+J,N
36 NEXT J
38 DATA 0,0,1,3,7,15,31,63
39 DATA 63,63,63,63,63,63,63,63
40 RW=6:CL=18
50 CLS
52 PRINT @2,6;" ";
60 INPUT "1 OR 2 PLAYERS";N
70 IF N<1 OR N>2 THEN 60
80 PRINT @2,10;"WHAT ARE THE PLAYERS NAMES?"
90 FOR J=1 TO N
100 PRINT @4,12+J*2;"PLAYER";J;
110 INPUT P$(J)
```

```

120 NEXT J
122 CLS
130 FOR L=1 TO N
140 A=INT(RND(1)*7)+4
150 B=INT(RND(1)*8)+3
160 C=INT(RND(1)*5)+3
162 IF C>=A THEN 140
172 SA=A*8
174 SB=B*8
176 SC=C*8
180 PRINT @CL,RW;CHR$(130);"a"
190 PRINT @CL-1,RW+1;CHR$(130);"ab"
200 PRINT @CL-2,RW+2;CHR$(130);"abb"
210 PRINT @CL-3,RW+3;CHR$(130);"abbb"
220 PRINT @CL-4,RW+4;CHR$(130);"abbbb"
222 PRINT @CL-5,RW+5;CHR$(130);"abbbbb"

230 FOR J=RW+6 TO RW+11
232 PRINT @CL-5,J;CHR$(135);CHR$(145);"
    ";CHR$(151)
234 NEXT J
240 PRINT @20,11;CHR$(128);A
250 PRINT @14,19;CHR$(128);B
260 PRINT @10,14;CHR$(128);C
270 PRINT @2,22;P$(L);", WHAT IS THE TO
TAL AREA";
280 INPUT ANS
284 GOSUB 700
290 IF ANS=C*B+B/2*(A-C) THEN 550
310 EXPLODE
320 PRINT @2,22;P$(L);", WHAT IS THE AR
EA OF THE"
322 PRINT "RECTANGLE";
330 INPUT RC
333 GOSUB 700
340 IF RC=C*B THEN 370
350 EXPLODE
360 PRINT @2,22;"THE AREA OF THE RECTAN
GLE IS";C*B
362 WAIT 500
363 GOSUB 700
364 GOTO 500
370 PRINT @2,22;P$(L);", WHAT IS THE AR
EA OF THE
    "
372 PRINT "TRIANGLE";

```




```

380 INPUT TG
384 GOSUB 700
390 IF TG=B/2*(A-C) THEN 420
400 EXPLODE
410 PRINT @2,22;"THE AREA OF THE TRIANG
LE IS";B/2*(A-C)
412 WAIT 500
414 GOTO 500
420 PRINT @2,22;P$(L);", WHAT IS THE TO
TAL AREA";
430 INPUT ANS
440 IF ANS=C*B+B/2*(A-C) THEN 470
450 EXPLODE
460 GOTO 500
470 PING
480 S(L)=S(L)+1
490 GOTO 562
500 GOSUB 700;PRINT @2,22;"THE TOTAL AR
EA IS";C*B+B/2*(A-C)
502 WAIT 200
506 CLS
510 GOTO 570
550 PING
560 S(L)=S(L)+2
562 CLS
570 PRINT @4,1;P$(1);" :";S(1);"      ";P
$(2);" :";S(2)
580 NEXT L
600 GOTO 130
700 FOR G=22 TO 24
710 PRINT @2,G;"
      "
720 NEXT G
730 RETURN

```

# 21

## Missing Letter



questions: 5  
player 1: 26 player 2: 33

saucr

missing letter?

correct!

If you are one of those poor souls who are always being told by someone that you miss letters in words don't worry too much because it's supposed to mean that you are clever and that your brain works faster than your hands. However, it's not a good idea to go on doing it so we have invented this game to help you.

Words are going to appear on the screen and you are going to have to spot the missing letter.

An example

sper

should be

spear

### How to play

Your computer will ask you 1 or 2 players?

## 88 *Missing Letter*

Type in 1 or 2 and press RETURN.

In the middle of the screen will appear a word with a letter missing.

To complete the word you press the letter of your choice.

It is **not necessary** to press the RETURN key

If your answer is correct the computer will tell you that your decision was correct, award you a number of points depending on the speed of your response and move on to the next word.

Wrong answers will mean no points and the computer will move on to the other player - if there are two playing.

### **Programming Notes**

It is possible to change the words used in the program by replacing those in the data statements of lines 350 to 440 with your own selection. If you increase the number of words though, change the 50 in lines 60, 90 and 190 to match the total number of words.

### **Program**

```
10 REM MISSING LETTER
20 CLS
30 PRINT:PRINT
40 INPUT "1 OR 2 PLAYERS";P
50 IF P<1 OR P>2 THEN 40
60 DIM W$(50)
70 CLS
82 IS=100
90 FOR J=1 TO 50
100 READ W$(J)
110 NEXT J
120 PRINT @9,1;CHR$(147);"QUESTIONS: 0"
;CHR$(151)
130 FOR J=1 TO P
140 PRINT @J*18-15,3;CHR$(147);"PLAYER
";J;" ":CHR$(151)
150 NEXTJ
```

```

160 Q=Q+1
170 FOR J=1 TO P
180 PRINT @J*18-15,3;CHR$(145);"PLAYER
";J;": ";S(J);CHR$(151)
190 R=INT(RND(1)*50)+1
200 R1=INT(RND(1)*5)+1
210 D$=LEFT$(W$(R),R1-1)+RIGHT$(W$(R),L
EN(W$(R))-R1)
214 PRINT @12,10;" "
220 PRINT @12,10;CHR$(150);D$;" ";CHR$(
151)
230 PRINT @9,16;CHR$(146);"MISSING LETT
ER?";CHR$(151)
240 A$=KEY$
250 IF A$="" THEN IS=IS-0.5:GOTO 240
260 IF IS<10 THEN IS=10
270 IF A$<>MID$(W$(R),R1,1) THEN 280
272 M$=" CORRECT! "
274 S(J)=S(J)+INT(IS/10)
275 IS=100
276 PING
278 GOTO 290
280 M$=" WRONG! "
282 EXPLODE
290 PRINT @11,20;CHR$(149);M$;CHR$(151)
300 PRINT @20,1;CHR$(147);Q;CHR$(151)
310 PRINT @J*18-15,3;CHR$(147);"PLAYER
";J;": ";S(J);CHR$(151)
320 WAIT 100
330 NEXT J
340 GOTO 160
350 DATA "HORSE","STREET","DAILY","MONE
Y","PEOPLE"
360 DATA "YELLOW","BOTTLE","ORANGE","SA
UCER","BREAKFAST"
370 DATA "COMPASS","KITTEN","ENGINE","H
ISTORY","SAUSAGE"
380 DATA "ADDRESS","CEILING","EXPERT","
MYSTERY","COMPUTER"
390 DATA "PROMISE","CLOWN","MESSAGE","E
ARLY","LISTEN"
400 DATA "MAGIC","SCHOOL","TOMORROW","K
ETTLE","DETACH"
410 DATA "MECHANIC","CYLINDER","SCIENTI
FIC","ILLOGICAL","GIRAFFE"

```


420 DATA "PARSLEY", "TRIANGLE", "LEGIBLE"  
,"MEDICINE", "PRESENCE"

430 DATA "TRIANGLE", "PENGUIN", "ALPHABET"  
,"DAFFODIL", "KNOWLEDGE"

440 DATA "YACHT", "FIERY", "KANGAROO", "RASPBERRY", "CHRISTMAS"

# 22

## Duck Shoot



Here is a game for your memory, your speed and your eyesight.

You will be shown a screen full of ducks - thirty in total - and then you will have to solve a multiplication sum before you can shoot at the sitting ducks.

Just in case you are worried about the ducks - they are not real, we borrowed them from a fairground.

### How to play

You will be asked to solve a multiplication table which will be selected at random by your computer. If you get the answer right and type in properly you will be allowed to shoot at the ducks.

## 92 *Duck Shoot*

It is not as simple as it sounds however because the gun moves from left to right and back again. You have to stop the gun which will fire on its own at the row of ducks. If you stop the gun in the wrong place the shot may go between a row of the 'quackers' and leave you without a score despite your correct answer to the sum.

To stop the gun you must press the space bar.

Your score board will show you how many shots were on target.

The trick is to be clever - and fast.

### Program

```
10 REM DUCK SHOOT
20 PAPER 0: INK 3
22 CLS
30 PRINT CHR$(17)
40 HT=0: SH=0
50 FOR J=0 TO 23
60 READ C
70 POKE 46856+J,C
80 NEXT J
90 DATA 0,12,30,31,63,7,3,1,0,0,0,63,63
,62,60,56
95 DATA 0,0,0,12,12,30,63,63
100 FOR J=2 TO 29 STEP 3
105 FOR K=6 TO 10 STEP 2
110 PRINT @J,K;CHR$(130);"ab";CHR$(144)
115 NEXT K
120 NEXT J
130 N1=INT(RND(1)*12)+1
140 N2=INT(RND(1)*12)+1
145 PRINT @8,17;"WHAT IS ";N1;" X ";N2;
" "
150 PRINT @2,23;"": INPUT ANS
154 PRINT @2,23;" "
160 IF ANS=N1*N2 THEN 200
165 EXPLODE
170 GOTO 150
200 CC=2
210 CD=1
215 PING
```

```


217 PRINT @8,17;"
"
220 PRINT @CC-1,20;" "
230 CC=CC+CD
240 IF CC>30 OR CC<2 THEN CD=-CD
250 PRINT @CC-1,20;CHR$(129);"c";CHR$(1
44)
260 IF KEY$<>" " THEN 220
270 SH=SH+1
275 ZAP
280 FOR J=19 TO 5 STEP -1
285 PLAY 1,0,0,0:MUSIC 1,3,INT(J/2),9:W
AIT 10:PLAY 0,0,0,0
290 IF PEEK(48040+CC+J*40)=97 OR PEEK(4
8040+CC+J*40)=98 THEN 330
300 PRINT @CC,J;". ";
303 WAIT 3
305 PRINT @CC,J;" ";
310 NEXT J
314 PRINT
320 GOTO 380
330 FOR B=1 TO 10
340 EXPLODE
350 NEXT B
360 PRINT @CC-1,J;" "
370 HT=HT+1
380 PRINT @4,1;CHR$(129);"SHOTS: ";SH;"
      HITS: ";HT;CHR$(144)
395 PRINT @CC,20;" "
400 IF HT=30 THEN END
410 GOTO 130

```



# 23

## Missing Numbers



A competition designed to see who is quickest between two players at working out the missing number. You can play on your own but it's no fun if there isn't anyone to show off to.

You'll be given a selection of sums and you will have to provide the number that has been replaced by a question mark.

Examples:

$$234 + ? = 563$$

$$? - 56 = 834$$

$$169 - 73 = ?$$

## How to play

You will be asked the players' names which you type in remembering to use capitals, and press RETURN

The computer will then ask you a question of the type shown above.

Type in your answer and press RETURN

Players will be asked questions alternately.

A wrong answer will mean a 'raspberry' buzz and the next question will move to your opponent.

Scores are shown on the screen.

## Program

```

10 REM MISSING NUMBER
20 CLS
30 PAPER 7:INK 4
40 PRINT@2,6;CHR$(131);CHR$(145);"WHAT
ARE THE PLAYERS NAMES?";CHR$(151)
50 PRINT
70 INPUT " PLAYER ONE ";P$(1)
80 PRINT
85 PING
90 INPUT " PLAYER TWO ";P$(2)
110 PING
120 WAIT 300
125 CLS
180 MAX=89
190 MIN=10
200 FOR C=1 TO 2
210 R(1)=INT(RND(1)*MAX)+MIN
220 R(2)=INT(RND(1)*MAX)+MIN
230 SIG=INT(RND(1)*2)
240 IF SIG=0 THEN SIG=-1
250 R(3)=R(1)+R(2)*SIG
255 IF R(3)<0 THEN 210
260 R=INT(RND(1)*3)+1
270 IF R=1 THEN PRINT@11,12; "??":GOTO
280

```

```


275 PRINT@11,12;RIGHT$(STR$(R(1)),2)
280 IF R=2 THEN PRINT@16,12; "??":GOTO
290
285 PRINT@16,12;RIGHT$(STR$(R(2)),2)
290 IF R=3 THEN PRINT@21,12; "?? ":GOTO
300
295 PRINT@21,12;RIGHT$(STR$(R(3)),LEN(S
TR$(R(3)))-1); " "
300 IF SIG<0 THEN PRINT@14,12; "-"
305 IF SIG>0 THEN PRINT@14,12; "+"
310 PRINT@19,12; "="
330 PRINT@2,18;CHR$(129);CHR$(147); "WH
AT IS THE MISSING NUMBER,";CHR$(151)
335 PRINT CHR$(129);CHR$(147);P$(C);CHR
$(151); " "
338 PRINT:PRINT
340 INPUT ANS
344 PRINT @2,22; " "
350 IF ANS<>R(R) THEN 430
360 PING
370 PRINT@12,8;CHR$(131);CHR$(145); "COR
RECT!";CHR$(151)
380 S(C)=S(C)+1
390 PRINT@3,1;CHR$(128);CHR$(146);P$(1)
; " ";S(1);CHR$(151)
400 PRINT@20,1;CHR$(128);CHR$(146);P$(2)
; " ";S(2);CHR$(151)
410 IF S(1)=20 OR S(2)=20 THEN 470
420 GOTO 445
430 EXPLODE
440 PRINT@12,6;CHR$(129);CHR$(147); "WRO
NG!";CHR$(151)
442 PRINT@4,8;CHR$(129);CHR$(147); "THE
MISSING NUMBER WAS ";R(R);CHR$(151)
445 WAIT 600
450 PRINT@10,6; " "
454 PRINT@4,8; "
"
456 PRINT@2,19; " "
460 NEXT C
465 GOTO 200
470 FOR W=1 TO 2
480 IF S(W)<>20 THEN 500
490 PRINT@5,5;CHR$(136);CHR$(146);CHR$(
27);"L";P$(W);" IS THE WINNER!";

```

```
495 PRINT CHR$(151)
500 NEXT W
510 FOR J=1 TO 17
520 PRINT
530 NEXT J
540 PRINT CHR$(17)
550 PAPER 7: INK 0
560 END
```

# 24

## Hangman



This is a game you may have played before using paper and pencil but now you are up against a computer.

The object of the game is to spell out a word correctly, in as few moves as possible, to save a man from being hanged. Every correct answer helps to complete the word, every wrong answer helps to hang the poor man.

You can, if you wish, change the title to hangwoman.

### How to play

The computer will put on the screen a word and will ask you for a letter to help fill in the blanks.

You don't need to press the RETURN key, just type the letter of your choice and the computer will do the rest.

Each letter you get right will be filled in but each letter you get wrong will help to complete the gallows, the rope and the victim.

The computer will also show the letters you have chosen incorrectly so that you don't choose them again and hang the man in error.

Scores are kept at the top of the screen.

Remember - think before you print - it could save a life.

### Programming Notes

The words for selection can be changed when typing in the data in lines 700 to 790. If you increase the number of words, change the 50 in lines 30, 50 and 80 to match the total number of words.

### Program

```

10 REM HANGMAN
20 INK 7:PAPER 0
22 CLS
30 DIM W$(50)
32 FOR J=0 TO 39
34 READ N
36 POKE 46856+J,N
38 NEXT J
40 DATA 12,12,12,12,12,12,12,12
42 DATA 63,63,0,0,0,0,0,0
44 DATA 1,3,6,12,24,48,32,0
46 DATA 32,48,24,12,6,3,1,0
48 DATA 12,30,63,45,63,33,30,12
50 FOR J=1 TO 50
60 READ W$(J)
70 NEXT J
80 R=INT(RND(1)*50)+1
82 LC=0
90 CT=LEN(W$(R))

```

```

92 PRINT @4,1;CHR$(129);"WORDS: ";NW;"
    SCORE: ";SC
150 FOR J=1 TO CT
160 PRINT @11+J,22;"-"
170 NEXT J
172 FOR C=8 TO 18
174 PRINT @15,C;CHR$(130)
176 NEXT C
178 FOR C=11 TO 15
180 PRINT @20,C;CHR$(129)
182 NEXT C
190 REM INPUT GUESS
200 G$=KEY$
205 IF G$<"A" OR G$>"Z" THEN 200
208 PING
210 FOR J=1 TO CT
220 IF G$=MID$(W$(R),J,1) THEN PRINT @1
1+J,22;G$:FL=1
240 NEXT J
241 C$=""
242 FOR J=1 TO LEN(W$(R))
243 T$=CHR$(PEEK(48040+22*40+11+J))
244 C$=C$+T$
245 NEXT J
246 IF C$=W$(R) THEN 470
249 IF FL=1 THEN FL=0:GOTO 190
250 PRINT @LC+4,20;G$
260 LC=LC+1
270 IF LC<>1 THEN 300
280 PRINT @16,18;"bbbbbb"
290 GOTO 190
300 IF LC<>2 THEN 310
302 FOR C=8 TO 17:PRINT @18,C;"a":NEXT
C
304 GOTO 190
310 IF LC<>3 THEN 320
312 PRINT @19,8;"bbbb"
314 GOTO 190
320 IF LC<>4 THEN 330
322 PRINT @23,8;"a"
323 PRINT @23,9;"a"
324 GOTO 190
330 IF LC<>5 THEN 340
332 PRINT @22,10;CHR$(131);"e"
334 GOTO 190
340 IF LC<>6 THEN 350

```

```

342 FOR C=11 TO 13:PRINT @23,C;"a":NEXT
C
344 GOTO 190
350 IF LC<>7 THEN 360
352 PRINT @22,11;"c"
353 PRINT @21,12;"c"
354 GOTO 190
360 IF LC<>8 THEN 370
362 PRINT @24,11;"d"
363 PRINT @25,12;"d"
364 GOTO 190
370 IF LC<>9 THEN 380
372 PRINT @22,14;"c"
373 PRINT @21,15;"c"
374 GOTO 190
380 IF LC<>10 THEN 460
382 PRINT @24,14;"d"
383 PRINT @25,15;"d"
384 GOTO 190
460 GOTO 540
470 FOR J=1 TO 12
480 MUSIC 1,3,J,10:PLAY 1,0,0,0:WAIT 20
:PLAY 0,0,0,0
490 NEXT J
510 WAIT 300
512 SC=SC+1
514 NW=NW+1
520 CLS
530 GOTO 80
540 FOR J=12 TO 1 STEP-1
550 MUSIC 1,3,J,10:PLAY 1,0,0,0:WAIT 20
:PLAY 0,0,0,0
560 NEXT J
580 WAIT 300
582 NW=NW+1
590 CLS
600 PRINT @8,10;CHR$(145);CHR$(128);"TH
E WORD WAS ";W$(R);CHR$(144)
610 WAIT 300
620 CLS
630 GOTO 80
700 DATA "SCHOOL","TOMORROW","ADDRESS",
"LENGTH","EXPERT"
710 DATA "KETTLE","COMPUTER","LEGIBLE",
"MEDICINE","BELIEVE"
720 DATA "MYSTERY","DOLPHIN","FEBRUARY"


```



, "BLANKET", "SPECIAL"  
730 DATA "PAVEMENT", "FRAGILE", "DIFFICUL  
T", "RHUBARB", "PENGUIN"  
740 DATA "ALPHABET", "DAFFODIL", "YACHT",  
"SQUARE", "THROUGH"  
750 DATA "TONGUE", "RASPBERRY", "CUSHION",  
"SOLDIER", "GENIUS"  
760 DATA "SPECTRUM", "CYLINDER", "OIL", "D  
ISPENSER", "SCATTER"  
770 DATA "CIRCUIT", "QUESTION", "DEFINITE",  
"PERCH", "CONSTRUCT"  
780 DATA "WEATHER", "DISTURB", "EXHAUST",  
"MOVEMENT", "SEMBLANCE"  
790 DATA "BANTAM", "OCTOBER", "SEPTEMBER",  
"ANNUAL", "PROPORTION"

# 25

## Organ



Here is your chance to become the next Stevie Wonder, Mozart or maybe just play 'Happy Birthday' to your parents on your computer.

You'll be able to type your own tunes into the computer's memory and have them played back to you.

### How to play

The top two lines on your computer, beginning Q and A are your keyboard and represent the keys of a piano. The top line beginning with Q represents the black keys and the bottom row beginning with A corresponds to the white keys.

When you press a key you will get a coloured symbol on the screen and a musical note from your computer.

When you have finished your tune you simply press the space bar and the computer will play back the whole tune for you.

If you want to clear the memory, and write a new tune, you simply press the letter X and the machine will be ready to take your next musical masterpiece.

### Program

```

10 REM ORGAN
30 DIM M(200),Q(200)
40 CLS
50 PRINT:PRINT "READY....":PRINT
60 A$=KEY$
70 IF A$="" THEN 60
72 O=3
80 IF A$="W" THEN N=2
90 IF A$="E" THEN N=4
100 IF A$="T" THEN N=7
110 IF A$="Y" THEN N=9
120 IF A$="U" THEN N=11
130 IF A$="O" THEN N=2:O=4
140 IF A$="P" THEN N=4:O=4
150 IF A$="A" THEN N=1
160 IF A$="S" THEN N=3
170 IF A$="D" THEN N=5
180 IF A$="F" THEN N=6
190 IF A$="G" THEN N=8
200 IF A$="H" THEN N=10
210 IF A$="J" THEN N=12
220 IF A$="K" THEN N=1:O=4
230 IF A$="L" THEN N=3:O=4
240 IF A$=" " THEN 318
250 IF A$="X" THEN C=0:CLS:GOTO 60
252 COL=INT(N/2)+145
260 MUSIC 1,O,N,O:PLAY 1,O,1,4000
262 PRINT CHR$(COL);" ";CHR$(151);
270 C=C+1
272 IF C>200 THEN C=200
280 M(C)=N:Q(C)=O


```

```
300 GOTO 60
318 CLS
320 FOR J=1 TO C
322 COL=INT(M(J)/2)+145
330 MUSIC 1,Q(J),M(J),0:PLAY 1,0,1,4000

332 PRINT CHR$(COL);" ";CHR$(151);
336 WAIT 30
340 NEXT J
350 GOTO 60
```

# 26

## Anagram



If you **think** that you are good at spelling then this game is for you. If you **know** that you are bad at spelling then this game is certainly for you. Well, you need the practice don't you? A selection of letters will be scattered around in a box within the screen and the player will be asked to unjumble them to make a word. Not any old word like MAFILY will do however, when the *correct* word is FAMILY.

### How to play

This is for one player.

The selection of letters will not be on one line but will be mixed up in a small square.

When you have decided what the correct word is you type it in and press RETURN.

Tries and correct answers will be shown on the screen but, in the event of a wrong answer the player will be asked to try again until the correct answer is given.

If you make a mistake press DELETE and begin again, but always remember to use CAPITAL letters when typing in your entry.

### Programming Hints

More difficult words can be used by changing the data statements in lines 430 to 470. If you increase the number of words, change the 20 in lines 30, and 100 to match the total number of words.

### Program

```

10 REM ANAGRAM
30 DIM W$(20)
40 PAPER 7:INK 0
60 CLS
70 FOR J=1 TO 20
80 READ W$(J)
90 NEXT J
92 GOTO 330
100 R=INT(RND(1)*20)+1
102 FOR J=7 TO 13
104 PRINT @11,J;CHR$(150);CHR$(128);"
    ";CHR$(151)
106 NEXT J
110 FOR J=1 TO LEN(W$(R))
130 P1=INT(RND(1)*6)+1
140 P2=INT(RND(1)*6)+1
150 IF M(P1,P2)=1 THEN 130
160 LET M(P1,P2)=1
170 PRINT @P2+11,P1+6;MID$(W$(R),J,1);
180 NEXT J
190 PRINT @6,16;CHR$(149);CHR$(128);" W
HAT IS THE WORD? ";CHR$(151)
194 PRINT:PRINT
200 FOR J=1 TO 6

```

```

210 FOR K=1 TO 6
220 M(J,K)=0
230 NEXT K
240 NEXT J
250 INPUT A$
256 PRINT @2,19;"
260 PRINT @6,16;"
      "

270 TR=TR+1
280 IF A$="QUIT" THEN 420
290 IF A$<>W$(R) THEN 390
300 PING
310 PRINT @11,16;CHR$(149);CHR$(128);"C
ORRECT!";CHR$(151)
320 SC=SC+1
330 PRINT @2,2;CHR$(147);CHR$(128);"TRI
ES: ";TR;CHR$(151)
340 PRINT @20,2;CHR$(147);CHR$(128);"SC
ORE: ";SC;CHR$(151)
342 WAIT 100
380 GOTO 100
390 EXPLODE
400 PRINT @10,16;CHR$(149);CHR$(128);"T
RY AGAIN!";CHR$(151)
405 WAIT 100
410 GOTO 190
420 PRINT @7,16;CHR$(149);"THE WORD WAS
";W$(R);" ";CHR$(151)
425 WAIT 300
428 PRINT @6,16;"
      "

430 GOTO 330
440 DATA "POND","WOOD","MOUSE","TIGER",
"DIGIT"
450 DATA "IDEA","ANAGRAM","MOTH","PARTY
","OCEAN"
460 DATA "PENNY","RABBIT","GUESS","PEAR
","TOAD"
470 DATA "BINARY","PUPIL","BASIC","VIDE
O","RECORD"

```

# 27

## Compass



This game, along with Morse Code and Constellation games' could save your life one day. You might wander off in a desert, a jungle or an ice field and you will thank your lucky stars that you once had a computer.

The Compass game will help you to identify North from South and even North East from South West.

### How to play

Your friendly computer will draw you a map and show you the eight major compass settings North, North East, East, South East, South, South West, West and North West.

The computer will then spin its own little compass setting, point the



arrow in a certain direction and ask you to decide where you are going.

Wrong answers mean that you will be asked again and again if necessary.

Random selection means that you may find yourself pointed in the same direction twice running but at least you should know where you are.

### Programming Notes

This program uses the same map routine as 'towns' and 'counties'. If you are typing in one of these as well you can save yourself some work if you just type in the lines that are different with your first program still in the computer. Don't forget to save your first program on a tape.

### Program

```

10 REM COMPASS
20 PAPER 0: INK 7
80 GOTO 140
100 CURSET 12,185,1
110 READ X,Y
111 IF X=-99 THEN RETURN
112 IF X>30 THEN CURSET X,Y,1:GOTO 110
120 DRAW X,Y,1
130 GOTO 110
140 INPUT "WHAT IS YOUR NAME";N$
142 IF LEN(N$)>8 THEN N$=LEFT$(N$,8)
146 HIRES
150 FOR J=1 TO 8
160 FOR K=1 TO 10
170 READ D(J,K)
180 NEXT K
190 READ T$(J)
200 NEXT J
202 GOSUB 100
210 FOR R=1 TO 8
220 C=1:GOSUB 500
240 PRINT T$(R)

```

```


244 PRINT
250 WAIT 250
260 C=0:GOSUB 500
270 NEXT R
290 R=INT(RND(1)*8)+1
300 C=1:GOSUB 500
316 PRINT "TRIES: ";TR
318 PRINT "SCORE: ";SC
320 INPUT "WHICH DIRECTION IS THIS";A$
328 TR=TR+1
330 IF A$=T$(R) THEN 400
340 EXPLODE
350 PRINT "TRY AGAIN, ";N$
360 WAIT 200
380 GOTO 316
400 PING
410 PRINT "CORRECT!"
420 WAIT 200
440 C=0:GOSUB 500
450 SC=SC+1
460 GOTO 290
500 FOR J=1 TO 10 STEP 2
510 X=D(R,J)
520 Y=D(R,J+1)
530 IF X>25 THEN CURSET X,Y,C:GOTO 550
540 DRAW X,Y,C
550 NEXT J
560 RETURN
700 DATA 115,85,0,-20,110,70,5,-5,5,5,"
NORTH"
710 DATA 105,85,20,-20,120,65,5,0,0,5,"
NORTH EAST"
720 DATA 105,75,20,0,120,70,5,5,-5,5,"E
AST"
730 DATA 105,65,20,20,125,80,0,5,-5,0,"
SOUTH EAST"
740 DATA 115,65,0,20,120,80,-5,5,-5,-5,
"SOUTH"
750 DATA 125,65,-20,20,105,80,0,5,5,0,"
SOUTH WEST"
760 DATA 125,75,-20,0,110,70,-5,5,5,5,"
WEST"
770 DATA 125,85,-20,-20,110,65,-5,0,0,5
,"NORTH WEST"
1000 DATA 5,-5,2,2,13,-9,1,-4,2,0,0,-2,
2,0,5,-10,3,-7,5,0,1,-5

```

1010 DATA 18,-3,12,2,3,-1,2,-5,5,-3,6,-  
 10,-10,7,-5,1,-3,4  
 1020 DATA -4,0,-9,-7,-9,0,0,-3,-5,-2,-9  
 ,5,-3,-4,3,0,0,-2,-7,0  
 1030 DATA 0,-2,3,0,0,-3,-3,0,5,-5,3,0,1  
 7,-15,2,-5,-2,-10,-2,-4  
 1040 DATA 1,-2,-10,-5,-3,-2,7,-7,13,-9,  
 2,2,7,-3,5,3,-3,-4  
 1050 DATA 6,4,5,-3,-6,0,-5,-7,3,-5,8,-3  
 ,-7,0,0,-8,8,-6,-4,2  
 1060 DATA 1,-4,-2,2,-2,-2,-5,4,-9,-15,4  
 ,-5,120,21  
 1070 DATA 1,2,-2,4,5,-1,4,0,12,13,-3,3,  
 3,8,4,3  
 1080 DATA 0,2,-8,-4,8,6,5,6,0,5,-5,6,8,  
 8,2,-9,6,-2  
 1090 DATA 7,2,8,4,3,12,-2,14,-9,14,-12,  
 5,5,0  
 1100 DATA 0,5,-10,6,5,1,-3,2,6,3,12,-2,  
 3,1,0,2,-4,1  
 1110 DATA 2,3,-5,6,-2,3,-4,-1,-10,12,-1  
 4,-4,-2,3  
 1120 DATA -10,1,0,1,-10,-4,-10,2,-9,1,2  
 ,2,-12,2,-9,-4  
 1130 DATA -8,3,-6,3,-3,12,-4,1,-8,-5,-7  
 ,1,-3,4,-1,-2  
 1140 DATA -2,2,0,4,-4,3,-6,-5,-6,3,-5,-  
 3,106,167,4,-2  
 1150 DATA 6,2,-4,4,-5,-4,155,135,4,0,0,  
 2,-2,0,-2,-2  
 1160 DATA 53,62,8,-5,-10,-5,-3,3,0,2,3,  
 6,-99,0

# 28

## Submarine



You are a destroyer captain alone in a hostile sea surrounded by a pack of submarines which are travelling secretly to a rendezvous. The submarines cannot break radio silence or send for help and must not attack you for fear of giving their position away but you can sink as many of them as you can - with as few depth charges as possible.

### How to play

On the screen will be shown a board divided into 100 squares. The submarine is hiding in one of those squares. The bottom (horizontal) line is called X and the upright (perpendicular) line is called Y.

Each line of boxes goes from 0 to 9 and you have to give the box numbers to the computer when it asks for your entry. You will be

asked to type in a number for the X and Y lines. If you think that the submarine is in a box 8 across and 5 high then type 8 and 5 when the X and Y positions are asked for.

**Remember** there is no need to press RETURN after each pair of numbers.

After the pair of numbers is entered you will hear the 'crump' of an exploding depth charge. If you make a direct hit first time you will hear a 'whooping' sound and the screen will show you how many tries you took to sink the submarine.

If however you miss, the computer will tell you if your shot was North, South, East or West of the target and you must then plan your next shot.

As soon as the submarine is sunk your computer will search and detect another target.

**Special Note** Expert captains should be able to detect and sink the enemy within four moves.

### Program

```

10 REM SUBMARINE
20 PRINT CHR$(17)
30 INK 7:PAPER 0
40 CLS
50 PRINT @3,0;CHR$(129);"SHOTS:";SH
60 PRINT @21,0;CHR$(129);"HITS:";HT
70 PRINT @8,2;"Y"
80 PRINT @22,16;"X"
90 FOR J=10 TO 19
100 PRINT @J,3;". "
110 PRINT @J,13;". "
120 PRINT @J,15;RIGHT$(STR$(J-10),1)
130 NEXT J
140 FOR J=3 TO 13
150 PRINT @9,J;". "
160 PRINT @20,J;". "
170 NEXT J
180 FOR J=4 TO 13
190 PRINT @7,J;STR$(13-J)

```

```

210 NEXT J
220 SY=INT(RND(1)*10)
230 SX=INT(RND(1)*10)
234 PRINT @2,23;"";
240 INPUT "YOUR SHOT (X,Y)";AX,AY
241 PRINT @2,23;"
      "
242 IF AX<0 OR AX>9 OR AY<0 OR AY>9 THE
N 240
245 PRINT @10+AX,13-AY;CHR$(145);CHR$(1
44)
250 FOR J=1 TO 10
260 MUSIC 1,3,11-J,10:PLAY 1,0,0,0:WAIT
10:PLAY 0,0,0,0
270 NEXT J
290 PRINT @2,18;"
      "
300 PRINT @2,18;
310 IF AX=SX AND AY=SY THEN 420
320 PRINT "YOUR SHOT WAS ";
330 IF AY<SY THEN PRINT "SOUTH ";
340 IF AY>SY THEN PRINT "NORTH ";
350 IF AX<SX THEN PRINT "WEST ";
360 IF AX>SX THEN PRINT "EAST ";
370 PRINT "OF THE"
374 PRINT "SUBMARINE, CAPTAIN."
380 SH=SH+1
390 PRINT @10,0;CHR$(129);SH
400 WAIT 200
410 GOTO 234
420 FOR J=1 TO 20
430 EXPLODE:WAIT 2
435 PRINT @10+AX,13-AY;CHR$(145);" ";CH
R$(144)
440 NEXT J
450 PRINT @2,18;"A DIRECT HIT!"
460 HT=HT+1
472 IF HT=10 THEN 540
480 WAIT 300
490 PRINT @2,18;"ANOTHER SUBMARINE HAS
BEEN DETECTED!"
500 WAIT 400
510 GOTO 40
540 FOR J=1 TO 10
550 MUSIC 1,3,J,10:PLAY 1,0,0,0:WAIT 10


```

116 *Submarine*

```
560 NEXT J
562 WAIT 50:PLAY 0,0,0,0
564 CLS
570 PRINT @11,8;"CONGRATULATIONS!"
580 PRINT @9,10;"TEN SUBMARINES SUNK!"
590 PRINT @11,13;"YOU HAVE BEEN"
600 PRINT @9,14;"PROMOTED TO ADMIRAL!"
```

# 29

## Word Search



This is a game to see how sharp your eyesight is and how quick you are at spelling words which are hidden in the screen.

The computer will put up a selection of letters all over the screen and, hidden amongst this alphabet spaghetti will be certain words which you will have to spot and spell out to your computer.

### How to play

When you find a word on the screen you type it in. You then press the RETURN key and, if you have correctly identified and spelled out the word, it will be picked out, in green, on your screen.

The score board will show your number of tries and the number of words you have correctly spotted.



When you cannot find any more words you simply type in the word QUIT.

Your computer will then blank out the words you have found and show, in red, all the words you have missed.

Your score will be shown on the screen.

### Programming Hints

If the selection of words is too simple or difficult you can change the data in lines 500 to 530. Make sure that all the words you use have exactly 7 letters.

### Program

```

10 REM WORD SEARCH
20 CLS
40 DIM M(20)
47 DIM B(17)
50 DIM W$(20)
52 DIM C$(20)
60 FOR J=1 TO 20
70 READ W$(J)
80 NEXT J
90 PRINT @2,3;
100 FOR K=1 TO 16
110 R1=INT(RND(1)*20)+1
115 IF M(R1)=1 THEN 110
117 M(R1)=1
119 C$(K)=W$(R1)
120 R2=INT(RND(1)*18)+3
125 B(K)=R2-1
130 FOR L=1 TO 32
140 IF L=R2 THEN PRINT C$(K);:L=L+7
150 LT=INT(RND(1)*26)+65
160 PRINT CHR$(LT);
170 NEXT L
180 PRINT
190 NEXT K
200 PRINT @2,23;""::INPUT "YOUR WORD";A$
$

```


```

202 IF A$="QUIT" THEN 400
204 PRINT @2,23;"
"
205 TR=TR+1
207 PRINT @11,21;"          "
210 FOR J=1 TO 16
220 IF A$=C$(J) THEN 270
230 NEXT J
240 EXPLODE
250 PRINT @12,21;CHR$(149);"WRONG!";CHR
$(151)
260 GOTO 300
270 PING
275 PRINT @B(J),J+2;CHR$(146);C$(J);CHR
$(151)
280 PRINT @11,21;CHR$(149);"CORRECT!";C
HR$(151)
285 C$(J)=" "
290 SC=SC+1
300 PRINT @4,0;CHR$(145);"TRIES: ";TR;C
HR$(151)
310 PRINT @20,0;CHR$(145);"WORDS: ";SC;
CHR$(151)
320 IF SC<>16 THEN 200
330 PRINT @2,24;" ";
332 END
400 FOR J=1 TO 16
404 IF C$(J)=" " THEN 420
410 PRINT @B(J),J+2;CHR$(145);C$(J);CHR
$(151)
420 NEXT J
430 GOTO 330
500 DATA "PROMISE","SPECIAL","BLANKET",
"FRAGILE","THROUGH"
510 DATA "ADDRESS","CEILING","AMATEUR",
"MYSTERY","LEGIBLE"
520 DATA "BELIEVE","STATION","RHUBARB",
"DISSECT","SAVIOUR"
530 DATA "ALCOHOL","CUSHION","SOLDIER",
"CABARET","AQUATIC"

```

# 30

## Noughts and Crosses



No, it's not that same old boring game that you play when it's raining outside and there is nothing else to do.

With our noughts and crosses you have to solve a problem **before** you can make your mark on your computer's board. You will have to decide if you can answer the question before you choose your box.

### How to play

The computer will ask for the players names which you type in and enter by pressing the RETURN key.

The computer will then present you with a board and in each of the nine squares you will find a maths sum to solve.

First choose the box you want and type the number.

Now you can type in your answer and see if you get your nought or cross for the correct answer.

Your opponent must follow the same steps.

If either player gives a wrong answer the computer moves to the opposing player's turn.

The game progresses until someone gets a winning line at which point you can press CTRL and C together, then RUN to play again.

If no one can complete a winning line then the game can proceed until all the boxes are completed and the player with the highest number of noughts or crosses will be the winner.

### **Programming Note**

You can make the sums easier by reducing the numbers in line 200 or more difficult by typing in a larger number.

### **Program**

```
10 REM NOUGHTS AND CROSSES
20 CLS
30 PAPER 7:INK 0
40 DIM N(18)
50 PRINT @2,6;"WHAT ARE THE PLAYERS NAMES?"
54 PRINT:PRINT
60 INPUT "PLAYER ONE";P$(1)
64 PRINT:PRINT
70 INPUT "PLAYER TWO";P$(2)
80 CLS
110 FOR J=2 TO 19
120 PRINT @11,J;CHR$(150);" ";CHR$(151)
130 PRINT @23,J;CHR$(150);" ";CHR$(151)
140 NEXT J
```

```

160 PRINT @3,6;CHR$(150);"
      ";CHR$(151)
170 PRINT @3,13;CHR$(150);"
      ";CHR$(151)
190 FOR J=1 TO 18
200 N(J)=INT(RND(1)*900)+100
210 NEXT J
220 C=1
230 FOR J=4 TO 28 STEP 12
240 FOR K=2 TO 16 STEP 7
250 PRINT @J,K;N(C)
260 C=C+1
270 PRINT @J,K+1;N(C)
280 C=C+1
290 PRINT @J+4,K;"+"
300 PRINT @J,K+3;"SQU";STR$(INT(C/2))
310 NEXT K
320 NEXT J
324 FOR N=1 TO 2
330 PRINT @2,21;P$(N);", WHICH NUMBER S
QUARE?"
340 INPUT SQ
342 PRINT @2,21;"SQUARE ";SQ;"
      "
344 IF SQ<1 OR SQ>9 THEN 330
345 IF H(SQ)=1 THEN 330
346 PRINT @2,21;"SQUARE ";SQ;"
      "
350 INPUT "WHAT IS THE ANSWER";ANS
354 PRINT @2,22;"
      "
360 IF ANS=N(SQ*2)+N(SQ*2-1) THEN 420
370 EXPLODE
400 GOTO 620
420 H(SQ)=1
422 PING
430 IF SQ>0 AND SQ<4 THEN W=4:V=SQ*7-5
440 IF SQ>3 AND SQ<7 THEN W=15:V=(SQ-3)
*7-5
450 IF SQ>6 AND SQ<10 THEN W=27:V=(SQ-6)
)*7-5
460 FOR J=0 TO 3
470 PRINT @W-1,J+V;"          "
480 NEXT J
490 IF N=2 THEN 560

```

```

500 PRINT @W+1,V+0;"X X"
510 PRINT @W+2,V+1;"X"
520 PRINT @W+1,V+2;"X X"
550 GOTO 620
560 PRINT @W+1,V+0;"000"
580 PRINT @W+1,V+1;"0 0"
600 PRINT @W+1,V+2;"000"
620 FA=0
622 FA=0
630 FOR J=1 TO 9
640 IF H(J)=0 THEN FA=1
650 NEXT J
654 IF FA=0 THEN 670
660 NEXT N
664 GOTO 324
670 PRINT @ 3,24;" "
680 END

```

# 31

## Towns



Would you believe that there are people living in this country who don't know where they are - on a map? Now we are sure that clever young readers of this book will have a good idea of where they live but will they know where other people live?

To find out we have devised this game to help you find out where the major cities in England and Wales are to be found. Scotland and Northern Ireland are on a separate program.

### How to play

The computer will draw an outline map for you and give you a list of ten towns to choose from.

The computer will then choose, at random, a town for you to identify. A bright dot will indicate the town in question.

Type in your answer in CAPITALS.

Tries and scores are shown on the screen.

If you are wrong it will be a case of trying until you get it right. After all, it pays to know where you are - doesn't it?

## Programming Notes

This program uses the same map drawing routine as 'counties' and 'compass'. If you are typing in all of them you will save yourself some work if you just type in the lines that are different with your first program still in the computer. Don't forget to save your first program onto a tape.

## Program

```

10 REM TOWNS
20 PAPER 0: INK 6
30 HIRES
50 FOR J=1 TO 10
60 READ M(J,1),M(J,2),T$(J)
70 NEXT J
80 GOTO 140
90 REM DRAW MAP
100 CURSET 12,185,1
110 READ X,Y
111 IF X=-99 THEN RETURN
112 IF X>30 THEN CURSET X,Y,1:GOTO 110
120 DRAW X,Y,1
130 GOTO 110
140 GOSUB 90
160 FOR K=1 TO 10
162 PRINT T$(K):PRINT
164 FOR C=1 TO 10
166 CURSET M(K,1),M(K,2),0:CIRCLE 3,1
174 WAIT 20
176 CURSET M(K,1),M(K,2),0:CIRCLE 3,0
177 WAIT 20

```



```

178 NEXT C
180 NEXT K
181 WAIT 200
182 R=INT(RND(1)*10)+1
184 FOR C=1 TO 5
190 CURSET M(R,1),M(R,2),0:CIRCLE 3,0
192 WAIT 10
194 CURSET M(R,1),M(R,2),0:CIRCLE 3,1
196 WAIT 10
198 NEXT C
200 PRINT "TRIES:";TR
210 PRINT "SCORE:";SC
220 INPUT "WHICH TOWN IS THIS";A$
230 CURSET M(R,1),M(R,2),0:CIRCLE 3,0
240 TR=TR+1
250 IF A$=T$(R) THEN 300
260 EXPLODE
270 PRINT "TRY AGAIN!"
274 WAIT 200
280 GOTO 184
300 PING
310 PRINT "CORRECT!"
320 SC=SC+1
334 WAIT 200
340 GOTO 182
900 DATA 143,158,"BRIGHTON"
910 DATA 97,137,"BRISTOL"
920 DATA 80,131,"CARDIFF"
930 DATA 114,94,"COVENTRY"
940 DATA 134,50,"HULL"
950 DATA 112,50,"LEEDS"
960 DATA 85,50,"LIVERPOOL"
970 DATA 145,137,"LONDON"
980 DATA 112,5,"NEWCASTLE"
990 DATA 52,175,"PLYMOUTH"
1000 DATA 5,-5,2,2,13,-9,1,-4,2,0,0,-2,
2,0,5,-10,3,-7,5,0,1,-5
1010 DATA 18,-3,12,2,3,-1,2,-5,5,-3,6,-
10,-10,7,-5,1,-3,4
1020 DATA -4,0,-9,-7,-9,0,0,-3,-5,-2,-9
,5,-3,-4,3,0,0,-2,-7,0
1030 DATA 0,-2,3,0,0,-3,-3,0,5,-5,3,0,1
7,-15,2,-5,-2,-10,-2,-4
1040 DATA 1,-2,-10,-5,-3,-2,7,-7,13,-9,
2,2,7,-3,5,3,-3,-4

```

```

1050 DATA 6,4,5,-3,-6,0,-5,-7,3,-5,8,-3
,-7,0,0,-8,8,-6,-4,2
1060 DATA 1,-4,-2,2,-2,-2,-5,4,-9,-15,4
,-5,120,21
1070 DATA 1,2,-2,4,5,-1,4,0,12,13,-3,3,
3,8,4,3
1080 DATA 0,2,-8,-4,8,6,5,6,0,5,-5,6,8,
8,2,-9,6,-2
1090 DATA 7,2,8,4,3,12,-2,14,-9,14,-12,
5,5,0
1100 DATA 0,5,-10,6,5,1,-3,2,6,3,12,-2,
3,1,0,2,-4,1
1110 DATA 2,3,-5,6,-2,3,-4,-1,-10,12,-1
4,-4,-2,3
1120 DATA -10,1,0,1,-10,-4,-10,2,-9,1,2
,2,-12,2,-9,-4
1130 DATA -8,3,-6,3,-3,12,-4,1,-8,-5,-7
,1,-3,4,-1,-2
1140 DATA -2,2,0,4,-4,3,-6,-5,-6,3,-5,-
3,106,167,4,-2
1150 DATA 6,2,-4,4,-5,-4,155,135,4,0,0,
2,-2,0,-2,-2
1160 DATA 53,62,8,-5,-10,-5,-3,3,0,2,3,
6,-99,0

```

```

10 REM TOWNS
12 REM SCOTLAND AND NI
20 PAPER 0:INK 6
30 HIRES
50 FOR J=1 TO 10
60 READ M(J,1),M(J,2),T$(J)
70 NEXT J
80 GOTO 140
90 REM DRAW MAP
100 CURSET 38,172,1
110 READ X,Y
111 IF X=-99 THEN RETURN
112 IF X>30 THEN CURSET X,Y,1:GOTO 110
120 DRAW X,Y,1
130 GOTO 110
140 GOSUB 90
160 FOR K=1 TO 10

```

```


162 PRINT T$(K):PRINT
164 FOR C=1 TO 10
166 CURSET M(K,1),M(K,2),0:CIRCLE 3,1
174 WAIT 20
176 CURSET M(K,1),M(K,2),0:CIRCLE 3,0
177 WAIT 20
178 NEXT C
180 NEXT K
181 WAIT 200
182 R=INT(RND(1)*10)+1
184 FOR C=1 TO 5
190 CURSET M(R,1),M(R,2),0:CIRCLE 3,0
192 WAIT 10
194 CURSET M(R,1),M(R,2),0:CIRCLE 3,1
196 WAIT 10
198 NEXT C
200 PRINT "TRIES: ";TR
210 PRINT "SCORE: ";SC
220 INPUT "WHICH TOWN IS THIS";A$
230 CURSET M(R,1),M(R,2),0:CIRCLE 3,0
240 TR=TR+1
250 IF A$=T$(R) THEN 300
260 EXPLODE
270 PRINT "TRY AGAIN!"
274 WAIT 200
280 GOTO 184
300 PING
310 PRINT "CORRECT!"
320 SC=SC+1
334 WAIT 200
340 GOTO 182
900 DATA 173,60,"ABERDEEN"
910 DATA 112,142,"AYR"
920 DATA 60,155,"BELFAST"
930 DATA 152,89,"DUNDEE"
940 DATA 147,120,"EDINBURGH"
950 DATA 124,125,"GLASGOW"
960 DATA 132,51,"INVERNESS"
970 DATA 59,145,"LARNE"
980 DATA 131,112,"STIRLING"
990 DATA 147,9,"THURSO"
1000 DATA -3,1,-4,-2,0,-2,-4,-3,-1,-4,-
2,-2,-2,3
1010 DATA -5,6,-2,3,-3,1,-3,-3,-4,-2,-1
,-3,-4,-1,1,-5,-1,-6,5,-2

```

1020 DATA 10,-3,-5,-4,0,-2,4,0,4,-4,7,-  
 10,3,1,3,-3,0,-4,3,2,6,-1  
 1030 DATA 3,-3,1,2,1,0,1,-1,4,3,5,-2,0,  
 8,3,0,-2,3,6,9  
 1040 DATA 1,-3,1,0,1,6,-7,8,9,-6,3,5,0,  
 3,2,2,0,6,-1,-3,-2,-3  
 1050 DATA -1,-2,-2,-2,-1,2,2,6,-3,3,5,-  
 1,1,3,-3,4  
 1060 DATA -2,-1,-2,2,-1,-3,-2,-3,-4,6,-  
 5,4,-8,-5,-4,2  
 1070 DATA 150,161,-15,2,0,2,-5,2,-2,3,-  
 3,0,-2,-2,-3,2,-4,-7  
 1080 DATA 0,11,-2,1,-3,-2,-2,-4,-5,-3,-  
 3,-4,-4,3,1,6,-1,1,-5,-13  
 1090 DATA 3,-3,1,4,1,0,0,-4,3,-5,3,-10,  
 3,-2,0,-3,-4,-5,1,-10,3,-3  
 1100 DATA 6,2,0,-1,-5,-10,-3,8,-2,-2,1,  
 -6,-2,0,-1,-6,-2,3,-2,-7,8,-7  
 1110 DATA -5,0,-6,5,0,10,-6,19,-3,0,-1,  
 -2,2,-2,0,-5,4,-5,-1,-13  
 1120 DATA 4,-5,-3,-4,13,-25,-1,0,-13,15,  
 -9,-5,0,-1,4,-3  
 1130 DATA 30,-70,6,3,1,4,6,-4,2,0,-2,3,  
 10,-4,4,1,10,1,4,-5  
 1140 DATA 6,4,-3,6,0,5,-20,15,0,3,-2,0,  
 0,2,5,-1,0,2,-5,4  
 1150 DATA -13,13,1,0,8,-7,1,1,17,-7,1,2,  
 6,2,4,-2,10,2,5,-1,6,2  
 1160 DATA 1,1,0,4,-2,5,-9,23,-5,5,1,2,-  
 5,6,-2,1,-4,-2  
 1170 DATA -8,4,-2,3,0,1,10,-4,-2,4,2,1,  
 3,3,-2,3,-8,0,-3,4  
 1180 DATA -2,1,-14,-4,-2,-1,0,1,11,7,5,  
 -1,4,3,2,0,4,-4,1,0,5,4  
 1190 DATA 6,0,1,1,5,1,3,4,0,4,2,3,-6,8,  
 -2,10,-20,14,-99,0

# 32

## Vowel Chase



You may have heard of voles and owls but we'll wager you haven't been on a vowel chase before.

You'll find yourself in the form of a little man running around an alphabetic maze collecting points for each vowel (A, E, I, O, U) which you munch.

But beware, there is a maze keeper out to munch you.

### How to play

Your tiny figure can be moved up and down using keys Q and A. For left and right use O and P respectively. Every time you munch a vowel you get one point and the space is filled in by your computer with a consonant, or even another vowel.

If you get all the vowels from the screen before you are caught you get 10 bonus points.

However, you also lose one point for every consonant you eat so try and stick to the maze corridors or you could wind up with a minus score.

As soon as the game is completed, or you have been munched up by the mazekeeper, your computer will ask you if you want to play again in which case you press Y or N for yes or no and press RETURN. The only advantage you have over the maze keeper is that you can run through the letters, which will cost you minus points, but the maze keeper must keep to the aisles.

If the maze keeper gets you and the game ends the screen will light up all the vowels you have missed.

## Program

```

10 REM VOWEL CHASE
20 CLS
24 PAPER 7::INK 0
30 POKE 590,5
40 PRINT CHR$(17)
50 FOR N=0 TO 15
60 READ D
70 POKE 46856+N,D
80 NEXT N
90 DATA 12,12,30,45,12,12,18,18
100 DATA 0,30,63,45,63,45,51,30
110 SC=0
112 PRINT @19,0;CHR$(149);"HI SCORE: ";
HS;CHR$(151)
120 PRINT @4,0;CHR$(149);"SCORE: ";CHR$(151)
130 PR=17
140 RP=17
150 PC=15
160 CP=15
170 GR=5
180 RG=5
190 GC=15

```

```

200 CG=15
206 FF=0
210 FOR J=4 TO 22 STEP 2
220 FOR K=4 TO 34 STEP 2
230 PRINT @K,J;CHR$(INT(RND(1)*26)+65)
240 NEXT K
250 NEXT J
260 PRINT @PC,PR;"a"
270 GOTO 970
300 A$=KEY$
310 IF A$="Q" AND PR>3 THEN PR=PR-1
320 IF A$="A" AND PR<23 THEN PR=PR+1
330 IF A$="O" AND PC>3 THEN PC=PC-1
340 IF A$="P" AND PC<35 THEN PC=PC+1
350 IF PC=CP AND PR=RP THEN 530
360 G$=CHR$(PEEK(48040+PC+PR*40))
364 IF G$="" OR G$=" " THEN 410
370 IF G$="A" OR G$="E" OR G$="I" OR G$
="O" OR G$="U" THEN 430
390 SC=SC-1
400 EXPLODE
410 GOTO 450
420 IF PEEK(48040+PC+PR*40)=98 THEN 800
424 GOTO 450
430 PING
440 SC=SC+1
450 PRINT @PC,PR;"a"
460 IF RP/2<>INT(RP/2) OR CP/2<>INT(CP/
2) THEN 480
470 PRINT @CP,RP;CHR$(INT(RND(1)*26)+65
):GOTO 490
480 PRINT @CP,RP;" "
490 RP=PR
500 CP=PC
510 PRINT @10,0;CHR$(149);" ";SC;" ";CH
R$(151)
530 RETURN
600 IF RND(1)>.7 THEN WAIT 10:GOTO 720
606 IF GR>PR THEN GR=GR-1:GOTO 650
610 IF GR<PR THEN GR=GR+1:GOTO 650
620 IF GC>PC THEN GC=GC-1:GOTO 650
630 IF GC<PC THEN GC=GC+1:GOTO 650
640 GR=GR+INT(RND(1)*3)-1:GC=GC+INT(RND
(1)*3)-1
650 IF PEEK(48040+GC+GR*40)=97 THEN 800

```

```


654 IF GR/2<>INT(GR/2) OR GC/2<>INT(GC/
2) THEN 660
656 GR=GR+INT(RND(1)*3)-1:GC=GC+INT(RND
(1)*3)-1
660 IF GR/2=INT(GR/2) AND GC/2=INT(GC/2
) THEN GR=RG:GC=CG:GOTO 720
670 PRINT @CG,RG;" "
680 PRINT @GC,GR;"b"
690 RG=GR
700 CG=GC
720 RETURN
800 PRINT @CG,RG;" "
810 PRINT @CP,RP;" "
820 FOR J=1 TO 10
824 WAIT 10
830 PRINT @GC,GR;"a"
840 ZAP:WAIT 10
850 PRINT @GC,GR;"b"
852 NEXT J
854 FOR J=2 TO 20 STEP 2
856 FOR K=2 TO 32 STEP 2
858 S$=CHR$(PEEK(48040+K+J*40))
860 IF S$<>"A" AND S$<>"E" AND S$<>"I"
AND S$<>"O" AND S$<>"U" THEN 864
862 FF=1:PRINT @K,J;CHR$(145);S$;CHR$(1
51)
864 NEXT K,J
865 IF FF=1 THEN 876
866 PRINT @10,10;CHR$(149);"BONUS 10 PO
INTS";CHR$(151)
872 SC=SC+10
874 PRINT @10,0;CHR$(149);" ";SC;" ";CH
R$(151)
876 PRINT @3,25;" "
878 INPUT "PLAY AGAIN";Q$
880 IF LEFT$(Q$,1)="N" THEN 1000
906 IF SC>HS THEN HS=SC
910 CLS
920 GOTO 110
970 GOSUB 300
974 A$=KEY$
980 GOSUB 600
990 GOTO 970
1000 END

```



# 33

## Parachute



Quick, quick, there is a plane crew parachuting down from a damaged airliner into a pine forest and the pilot's parachute hasn't opened - only you can save him from crashing into the ground.

### How to play

At the top of the screen you will see the height and the speed of the pilot's descent. To open the parachute you must divide the height by the speed to find out how long it will take for the pilot to land. If your answer is correct the parachute will open and your pilot will float gently to the ground.

Your crew have three lives between them so try and keep them alive as long as you can.

The sooner you get the answer and type it in and press RETURN the earlier the parachute will open and the more points you will receive.

## Programming Hints

You can make the game easier by reducing the numbers in lines 210 and 220.

## Program

```

10 REM PARACHUTE
20 PRINT CHR$(17)
40 LV=3
80 TR=0
84 PAPER 0: INK 4
86 CLS
90 FOR J=0 TO 63
100 READ N
110 POKE 46856+J,N
120 NEXT J
130 DATA 7,15,31,63,63,37,19,9
140 DATA 48,56,60,62,62,10,20,40
150 DATA 5,3,1,7,11,3,2,4
160 DATA 16,32,0,48,40,32,32,16
170 DATA 8,28,9,30,40,28,20,34
180 DATA 12,12,12,12,12,30,0,0
190 DATA 0,12,30,30,63,63,63,63
200 DATA 0,0,0,33,30,62,31,42
210 SP=INT(RND(1)*90)+9
220 ANS=INT(RND(1)*12)+1
230 HT=ANS*SP
240 PRINT @2,0;CHR$(147);"HEIGHT:";HT;C
HR$(144)
250 PRINT @16,0;CHR$(147);"SPEED:";SP;C
HR$(144)
260 PRINT @29,0;CHR$(147);"LIVES:";LV;C
HR$(144)
270 PRINT @3,2;CHR$(147);"SCORE:";SC;"
";CHR$(144)
280 PRINT @21,2;CHR$(147);"HI SCORE:";
HS;CHR$(144)
290 PRINT @2,6;CHR$(146);"TIME ?? ";CHR
$(144)

```

```

294 PRINT @ 2,20;CHR$(130)
296 PRINT @ 2,21;CHR$(129)
300 FOR J=4 TO 30
310 IF J>12 AND J<20 THEN 340
320 PRINT @J,21;"f"
330 PRINT @J,20;"g"
340 NEXT J
350 FOR D=4 TO 20
360 PRINT @15,D-1;" "
370 MUSIC 1,3,12-D/2,9:GOSUB 1000
380 PRINT @15,D;CHR$(134);"e"
390 FOR T=1 TO 80
400 A$=KEY$
410 IF A$="" THEN 460
420 IF A$=CHR$(13) THEN 560
430 IF A$>"9" OR A$<"0" THEN 460
440 N$=N$+A$
442 FOR L=1 TO 50:NEXT L
450 PRINT @8,6;CHR$(146);N$;" ";CHR$(14
4)
460 NEXT T
470 NEXT D
480 PRINT @15,20;" "
490 MUSIC 1,2,4,8:GOSUB 1000
500 PRINT @15,21;CHR$(129);"h"
510 LV=LV-1
520 FOR L=1 TO 1000:NEXT L
530 IF LV=0 THEN 790
540 PRINT @15,21;" "
550 GOTO 210
560 MUSIC 1,4,2,9:GOSUB 1000
562 IF N$="" THEN 460
564 TR=TR+3
570 CA=VAL(N$)
580 IF CA=ANS THEN 640
590 MUSIC 1,2,4,8:GOSUB 1000
600 PRINT @2,6;CHR$(146);"TIME ?? ";CHR
$(144)
610 N$=""
620 GOTO 460
640 SC=SC+23-D-TR
650 FOR J=D TO 20
660 PRINT @15,J-2;" "
670 PRINT @15,J-1;" "
680 PRINT @15,J;CHR$(131);"a"+"b"
690 PRINT @15,J+1;CHR$(134);"c"+"d"

```

```

700 MUSIC 1,3,J/2,9:GOSUB 1000
710 WAIT 5
720 NEXT J
722 PRINT @ 18,20;CHR$(130):PRINT @ 18,
21;CHR$(129)
724 WAIT 25
730 MUSIC 1,3,8,8:GOSUB 1000
740 MUSIC 1,3,12,9:GOSUB 1000
742 PRINT @6,12;CHR$(149);"A SUCCESSFUL
LANDING!";CHR$(144)
744 PRINT @8,14;CHR$(149);"SCORE ";23-D
-TR;" POINTS";CHR$(144)
750 WAIT 250
760 PRINT @15,20;"  "
770 PRINT @15,21;"  "
772 PRINT @6,12;"
"
774 PRINT @8,14;"
"
776 N$=""
778 TR=0
780 GOTO 210
790 MUSIC 1,3,2,8:GOSUB 1000
800 MUSIC 1,2,8,9:GOSUB 1000
810 CLS
820 PRINT @7,8;CHR$(149);"YOUR SCORE IS
";SC;CHR$(144)
830 IF SC<HS THEN 850
832 HS=SC
840 PRINT @7,12;CHR$(149);"A NEW HIGH S
CORE!";CHR$(144)
850 PRINT @10,18;CHR$(146);"PLAY AGAIN?
";CHR$(144)
860 Q$=KEY$
870 IF Q$="" THEN 860
880 IF LEFT$(Q$,1)="N" THEN 930
890 SC=0
900 LV=3
904 N$=""
906 TR=0
910 CLS
920 GOTO 210
930 PAPER 7:INK 0
940 CLS
950 END

```

138 *Parachute*

```
1000 PLAY 1,0,0,0:WAIT 25:PLAY 0,0,0,0:  
RETURN  
1900 ,12,30,30,63,63,63,63  
18012 ,12,12,12,12,30,0,0
```

# 34

## Racer



It's the last Grand Prix of the year and there are only two drivers who can win the World Championship - you or your opponent. There are two cars to choose from and the finishing line is in sight.

Only one can win and it's done by brain power not horse power.

### How to play

The game is for two players so firstly type in your names and press RETURN after each one.

Both cars appear on the screen and the first player is set a problem to solve before moving their car.

The question is 'What is your estimate of the distance travelled?' At the top of the screen you are shown the speed and the time taken by the car. You must work out how far you have come. The more accurate your answer the further your car will be moved by the computer so think carefully before you type in your answer.

The winner's name will be flashed on the screen and you will be asked 'do you want to play again'. Press Y or N for yes or no.

**Hint** To help you, 36 kilometres per hour = 10 metres per second.  
So a car travelling at 72 k. per hour for 20 seconds would cover 400 metres.

### Program

```

10 REM RACER
18 PAPER 7: INK 0
20 CLS
30 PRINT CHR$(17)
80 FOR N=0 TO 15
90 READ D
100 POKE 46856+N,D
110 NEXT N
120 DATA 0,0,0,0,54,63,63,54
130 DATA 51,51,12,12,51,51,12,12
140 PRINT @2,6;CHR$(149);"WHAT ARE THE
PLAYERS NAMES?";CHR$(151)
150 PRINT
160 INPUT P$(1)
170 PRINT @2,10;CHR$(145);P$(1);" YOU H
AVE THE RED CAR";CHR$(151)
180 PRINT
190 INPUT P$(2)
200 PRINT @2,14;CHR$(146);P$(2);" YOU H
AVE THE GREEN CAR";CHR$(151)
210 WAIT 300
220 CLS
250 PRINT @2,6;CHR$(150);"
";CHR$(151)
260 PRINT @2,12;CHR$(150);"
";CHR$(151)
280 FOR J=7 TO 11
290 PRINT @30,J;CHR$(129);"b"

```

```

292 NEXT J
300 N(1)=1
310 N(2)=1
320 PRINT @2,8;CHR$(129);"a"
330 PRINT @2,10;CHR$(130);"a"
340 FOR K=1 TO 2
350 TM=INT(RND(1)*90)+10
360 SP=INT(RND(1)*150)+50
362 PRINT @2,1;"
      "
370 PRINT @2,1;CHR$(147);"SPEED:";SP;"K
M/HR ";CHR$(151)
380 PRINT @20,1;CHR$(147);"TIME:";TM;"
SEC ";CHR$(151)
390 PRINT @2,20;CHR$(K+144);P$(K);
392 PRINT " WHAT IS YOUR";CHR$(151)
394 PRINT CHR$(K+144);"ESTIMATE OF THE
DISTANCE TRAVELLED?";CHR$(151)
398 PRINT @2,23;" ";
400 INPUT "NUMBER OF METRES";ANS
404 PRINT @2,23;"
      "
406 PRINT @2,20;"
      "
410 CA=(SP*TM*1000)/3600
420 SC=INT(5-ABS((CA-ANS)/50))
430 FOR J=N(K) TO N(K)+SC
440 PRINT @N(K),6+K*2;" "
450 N(K)=N(K)+1
460 PRINT @N(K),6+K*2;CHR$(K+128);"a"
470 PLAY 1,0,0,0:MUSIC 1,3,7,8:WAIT 10:
PLAY 0,0,0,0
480 IF N(K)=30 THEN 520
490 NEXT J
500 NEXT K
510 GOTO 340
520 PLAY 1,0,0,0:MUSIC 1,3,4,8:WAIT 50
530 MUSIC 1,3,9,10:WAIT 100:PLAY 0,0,0,
0
540 WAIT 200
550 CLS
560 PRINT @6,8;CHR$(K+144);P$(K);" IS T
HE WINNER!";CHR$(151)
570 PRINT @9,14;CHR$(149);"PLAY AGAIN?"
;CHR$(151)

```



142 *Racer*

```
580 A$=KEY$  
590 IF A$="" THEN 580  
600 IF LEFT$(A$,1)="Y" THEN 220  
610 END
```

# 35

## Proverbs



There is an old proverb that says 'every cloud has a silver lining' and another which says 'it's an ill wind that blows nobody any good'.

But if your proverbs read 'every silver cloud has a lining' or 'it's nobody that an ill wind blows any good' could you sort them out and rearrange them properly?

Well, here is your chance to use the edit and cursor keys on your computer to move words around.

### How to play

The screen will show you a proverb all jumbled up and you must reposition the words to make sense using the arrow keys to move

the cursor up and down.

Key D to DELETE

Key I to INSERT

Line the cursor up against the word you wish to move and press the D key. The word will then vanish and you move the cursor down the board until you line up against the word where you wish to insert.

Press I and the missing word will be pushed in and the other words will rearrange below.

You keep moving words and using the D and I keys until you have the proverb in the correct order.

### **Programming Hints**

As the program is random you may get the same proverb twice running. You can change lines 900 to 990 to put in your own proverbs or well known sayings.

### **Program**

```

10 REM PROVERBS
20 CLS
40 DIM W$(10,12)
42 DIM S(12)
44 DIM Q(12)
50 C=1
52 TM=1
54 MM=0
56 FF=0
60 FOR J=1 TO 10
70 READ T$
80 FOR K=2 TO LEN(T$)
90 IF MID$(T$,K,1)<>" " THEN 130
100 W$(J,C)=MID$(T$,TM+1,K-TM)
110 TM=K
120 C=C+1

```

```
130 NEXT K
140 C=1
150 TM=1
160 NEXT J
180 WN=INT(RND(1)*10)+1
190 WC=0
200 FOR J=1 TO 12
210 IF W$(WN,J)="" THEN 230
220 WC=WC+1
230 NEXT J
250 FOR J=1 TO WC
260 R=INT(RND(1)*WC)+1
262 IF Q(R)=1 THEN 260
270 S(J)=R
272 Q(R)=1
280 PRINT @5,J+4;W$(WN,S(J))
290 NEXT J
300 PRINT @4,5;">"
310 CP=5
320 WAIT 10
322 G$=KEY$
330 IF G$="" THEN 320
340 IF G$(<>CHR$(10)) THEN 400
350 PRINT @4,CP;" "
360 CP=CP+1
370 IF CP>20 THEN CP=CP-1
380 PRINT @4,CP;">"
390 GOTO 320
400 IF G$(<>CHR$(11)) THEN 460
410 PRINT @4,CP;" "
420 CP=CP-1
430 IF CP<1 THEN CP=CP+1
440 PRINT @4,CP;">"
450 GOTO 320
460 IF G$(<>"D") THEN 520
470 IF CP<5 OR CP>5+WC THEN 320
480 IF MM<>0 THEN ZAP:GOTO 320
490 MM=S(CP-4)
492 FOR K=CP-4 TO 11
494 S(K)=S(K+1)
496 NEXT K
500 CLS
510 GOTO 590
520 IF G$(<>"I") THEN 320
522 IF MM=0 THEN 320
```


```

530 IF CP<5 OR CP>5+WC THEN 320
540 FOR J=12 TO CP-4 STEP -1
542 IF J<=1 THEN 560
550 S(J)=S(J-1)
560 NEXT J
570 S(CP-4)=MM
580 CLS
582 MM=0
590 FOR J=1 TO 12
592 IF S(J)=0 THEN 620
600 PRINT @5,J+4;W$(WN,S(J))
610 IF S(J)<>J THEN FF=1
620 NEXT J
630 IF FF=0 THEN 700
640 FF=0
650 GOTO 380
700 PING
710 CLS
712 FOR J=1 TO 12
714 PRINT @5,J+4;W$(WN,J)
716 NEXT J
720 PRINT @2,21;"THAT IS THE CORRECT OR
DER"
730 WAIT 300
740 FOR J=1 TO 12
750 S(J)=0
760 Q(J)=0
770 NEXT J
774 MM=0
780 CLS
790 GOTO 180
900 DATA " MANY HANDS MAKE LIGHT WORK "
910 DATA " TOO MANY COOKS SPOIL THE BRO
TH "
920 DATA " A STITCH IN TIME SAVES NINE
"
930 DATA " EVERY CLOUD HAS A SILVER LIN
ING "
940 DATA " DON'T PUT ALL YOUR EGGS IN O
NE BASKET "
950 DATA " LOOK BEFORE YOU LEAP "
960 DATA " MORE HASTE LESS SPEED "
970 DATA " DON'T CROSS BRIDGES UNTIL YO
U COME TO THEM "
980 DATA " PRIDE COMES BEFORE A FALL "
990 DATA " HE WHO RIDES THE TIGER CANNO
T DISMOUNT "

```

# 36

## Simon



This is our version of this very popular game where you have to beat the computer by having a good memory.

You will be shown numbers, or colours, and have only a few seconds to memorise them before the screen goes blank and you are asked to repeat them in the same sequence as the original.

### How to play

Your computer will ask if you want to play Numbers or Colours and you make your selection using keys 1 or 2. The computer will then put a sequence of numbers on the screen and ask you to remember them. If you type in the wrong sequence the computer will give you a loud 'raspberry' and show you the correct answer.

To stop playing numbers and switch to colours press the CTRL and C key and then type RUN again and you will be back at the beginning.

To play the colour sequence you have to press number keys corresponding to the Atmos colour. A list of numbers and colours can be found in your Atmos Handbook, or it may be worth listing them in the book:

- 0 Black
- 1 Red
- 2 Green
- 3 Yellow
- 4 Blue
- 5 Magenta
- 6 Cyan
- 7 White

## Programming Hints

The two routines for selecting the numbers and colours are in lines 210 and 250. Why not try and write your own routines to generate letters as well? Remember that computers recognise letters by their 'ASCII' codes which start with 65 for A, then 66 for B and so on.

## Program

```

10 REM SIMON
20 CLS
30 PAPER 7:INK 0
40 T=1
70 PRINT @4,6;CHR$(146);"1. NUMBERS";CHR$(151)
80 PRINT @4,8;CHR$(146);"2. COLOURS";CHR$(151)
90 PRINT @4,12;CHR$(146);"PRESS 1 OR 2";CHR$(151)
100 A$=KEY$
110 IF A$="" THEN 100
112 PING

```

```

120 IF A$<>"1" AND A$<>"2" THEN 100
130 M$=""
140 CLS
150 PRINT @4,0;CHR$(147);"SEQUENCES:";T
R;CHR$(151)
160 PRINT @22,0;CHR$(147);"SCORE:";SC;C
HR$(151)
170 PRINT @6,6;
180 C=INT(RND(1)*6)+144
190 FOR J=1 TO T
200 IF A$="2" THEN 250
210 R=INT(RND(1)*10)
220 PRINT RIGHT$(STR$(R),1);
240 GOTO 270
250 R=INT(RND(1)*7)
260 PRINT CHR$(144+R);" ";CHR$(151);
270 M$=M$+RIGHT$(STR$(R),1)
280 NEXT J
282 WAIT 500
284 PRINT @6,6;"
"
290 PRINT @6,16;CHR$(146);"WHAT IS YOUR
ANSWER?";CHR$(151)
294 PRINT:PRINT
300 INPUT N$
310 TR=TR+1
320 IF N$=M$ THEN 440
330 EXPLODE
340 CLS
350 PRINT @4,6;CHR$(149);"THE SEQUENCE
WAS:";CHR$(151)
360 PRINT @6,10;
370 IF A$="2" THEN 400
380 PRINT M$
390 GOTO 424
400 FOR J=1 TO T
410 PRINT CHR$(144+VAL(MID$(M$,J,1)));
";CHR$(151);
420 NEXT J
424 WAIT 400
430 GOTO 130
440 PING
450 PRINT @11,18;CHR$(146);"CORRECT!";C
HR$(151)
460 SC=SC+T

```



150 *Simon*


470 T=T+1

474 WAIT 150

480 GOTO 130

# 37

## Balance



If you like baking cakes, weighing out sweets or even checking out who has the heaviest conker this year then this program is for you.

You have to work out, in as few moves as possible, from a selection of weights, how many you need to balance a set of scales.

### How to play

On the screen you will see a set of scales with a weight on the left hand set of pans.

The computer will ask you for your choice of weights from six which will be displayed, and marked, along the bottom of the screen. The computer will move along the row and stop at each

## 152 *Balance*

weight and ask you if you wish to use it. For yes press Y and for no press N.

The screen will show your tries and scores.

### Program

```
10 REM BALANCE
14 PAPER 7:INK 0
20 CLS
22 FOR J=0 TO 39
24 READ N
26 POKE 46856+J,N
28 NEXT J
30 DATA 12,12,12,12,12,12,12,12
32 DATA 63,63,0,0,0,0,0,0
34 DATA 1,3,6,12,24,48,32,0
36 DATA 32,48,24,12,6,3,1,0
38 DATA 0,0,0,0,0,0,63,63
40 K=25:CLS
42 PRINT @4,0;CHR$(147);"TRIES:";TR;CHR$
(151)
44 PRINT @20,0;CHR$(147);"SCORE:";SC;CH
R$(151)
46 PRINT @2,17;CHR$(132)
50 FOR J=3 TO 36 STEP 6
60 PRINT @J,17;STR$(K);"g"
70 K=K*2
80 NEXT J
90 GOSUB 900
100 IC=INT(RND(1)*5)+2
110 TT=0
120 R=(INT(RND(1)*63)+1)*25
130 PRINT @2,2;CHR$(146);"YOUR WEIGHT T
O BALANCE IS ";R;"g.";CHR$(151)
140 TR=TR+1
170 PRINT @16,14;CHR$(IC+144);" ";CHR$
(151)
180 K=25
190 FOR J=1 TO 6
200 PRINT @4,21;CHR$(146);"DO YOU WANT
A";STR$(K);"g WEIGHT?";CHR$(151)
202 A$=KEY$
```

```

204 IF A$="" THEN 202
206 IF A$="Y" THEN TT=TT+K:PRINT @(J-1)
*6+4,19;CHR$(149);"1";CHR$(151)
208 IF A$<>"Y" THEN PRINT @(J-1)*6+4,19
;CHR$(147);"0";CHR$(151)
210 FOR D=1 TO 50:NEXT D
214 K=K*2
216 PING
220 NEXT J
222 PRINT @25,6;CHR$(145);" ";CHR$(151)
)
226 WAIT 100
228 IF TT<R THEN 330
230 GOSUB 600
250 IF TT>R THEN 300
260 GOSUB 1100
262 PRINT @16,10;CHR$(IC+144);" ";CHR$(151)
264 PRINT @24,10;CHR$(145);" ";CHR$(151)
1)
270 PLAY 1,0,0,0:MUSIC 1,3,8,8:WAIT 50:
MUSIC 1,4,1,12
274 WAIT 70:PLAY 0,0,0,0
280 SC=SC+1
290 GOTO 340
300 GOSUB 1000
310 PRINT @15,6;CHR$(144+IC);" ";CHR$(151)
320 PRINT @25,14;CHR$(145);" ";CHR$(151)
1)
330 EXPLODE
340 WAIT 600
350 GOTO 40
600 FOR C=3 TO 15
610 PRINT @4,C;"
"
620 NEXT C
630 RETURN
900 PRINT @25,3;"ca"
905 PRINT @24,4;"c a"
910 PRINT @23,5;"c a"
915 PRINT @22,6;"c eee"
920 PRINT @21,7;"ca"
925 PRINT @20,8;"c a"
930 PRINT @19,9;"c a"

```


```

935 PRINT @18,10;"c    a"
940 PRINT @17,11;"a    a"
945 PRINT @17,12;"a    a"
950 PRINT @17,13;"a    a"
955 PRINT @16,14;"eee  a"
960 PRINT @15,15;"bbbbbbbbbbbbbbb"
965 RETURN
1000 PRINT @17,3;"ad"
1005 PRINT @17,4;"a d"
1010 PRINT @17,5;"a  d"
1015 PRINT @16,6;"eee  d"
1020 PRINT @21,7;"ad"
1025 PRINT @21,8;"a d"
1030 PRINT @21,9;"a  d"
1035 PRINT @21,10;"a   d"
1040 PRINT @21,11;"a    a"
1045 PRINT @21,12;"a    a"
1050 PRINT @21,13;"a    a"
1055 PRINT @21,14;"a    eee"
1060 PRINT @15,15;"bbbbbbbbbbbbbbb"
1065 RETURN
1100 PRINT @18,6;"eeeeeee"
1105 PRINT @17,7;"a    a    a"
1110 PRINT @17,8;"a    a    a"
1115 PRINT @17,9;"a    a    a"
1120 PRINT @16,10;"eee  a  eee"
1125 PRINT @21,11;"a"
1130 PRINT @21,12;"a"
1135 PRINT @21,13;"a"
1140 PRINT @21,14;"a"
1145 PRINT @15,15;"bbbbbbbbbbbbbbb"
1150 RETURN

```

# 38

## Chemist



A mad scientist has made a highly powerful liquid which, if not mixed with water and diluted properly, will explode and destroy your house.

Only you can stop the liquid exploding by working out a formula to decide how much water is needed to make the mixture safe.

### How to play

The screen will show a flask containing the dangerous fluid and telling you how many litres the vessel contains.

You must work out how much water is required to make the fluid harmless using the ratio at the top of the screen. If you are correct to within 5% you are safe but an error outside this range will mean an explosion and the loss of one of your three lives.

Type in your estimate of water needed, press RETURN and watch the water drip into the flask - Good Luck.

### Programming Hints

You can alter the ratios given by changing the numbers in lines 60 and 64. Increase them to make the game more difficult. The number in line 50 governs the number of lives you have at the start.

### Program

```

10 REM CHEMIST
20 CLS
30 PAPER 7: INK 0
40 PRINT CHR$(17)
50 LV=3
60 A=INT(RND(1)*4)+1
64 B=INT(RND(1)*5)+5
68 PRINT @8,6;CHR$(145);"YOUR RATIO IS:
";CHR$(151)
70 PRINT @2,8;CHR$(145);A;" PARTS FLUID
TO ";B;" PARTS WATER";CHR$(151)
74 WAIT 500
80 CLS
90 PRINT @4,0;CHR$(147);"LIVES LEFT:";L
V;CHR$(151)
100 PRINT @23,0;CHR$(147);"SCORE:";SC;C
HR$(151)
102 PRINT @8,3;CHR$(145);"YOUR RATIO IS
";A;" ":";B;CHR$(151)
150 R=INT(RND(1)*20)+1
160 PRINT @4,6;CHR$(146);"YOU HAVE ";R;
" LITRES OF FLUID";CHR$(151)
170 PRINT @3,8;CHR$(146);"HOW MUCH WATE
R WILL YOU USE?";CHR$(151)
180 PRINT @14,16;CHR$(145);"      ";CHR$(
151)
184 PRINT @14,17;CHR$(145);"      ";CHR$(
151)
188 PRINT @2,22;" ";
190 INPUT ANS
196 PRINT @7,20;CHR$(146);ANS;"LITRES O
F WATER";CHR$(151)

```

```

200 T=(R/A)*B
210 FOR J=1 TO 3
220 FOR K=10 TO 15
230 PRINT @16,K;CHR$(150);" ";CHR$(151)
234 WAIT 2
240 PRINT @16,K;" "
250 MUSIC 1,3,K-5,10:PLAY 1,0,0,0:WAIT
10:PLAY 0,0,0,0
260 NEXT K
270 NEXT J
280 FOR J=15 TO 12 STEP -1
290 PRINT @14,J;CHR$(150);"      ";CHR$(1
51)
300 NEXT J
310 WAIT 100
320 IF ANS>T-T/20 AND ANS<T+T/20 THEN 5
00
326 CLS
330 FOR J=6 TO 1 STEP -1
340 PAPER J
350 EXPLODE
354 WAIT 50
360 CLS
370 NEXT J
374 PAPER 7
380 PRINT @9,6;CHR$(151);"ONE LIFE LOST
!"
390 WAIT 250
400 LV=LV-1
410 PAPER 7
420 IF LV=0 THEN 590
430 GOTO 80
500 PING
510 PING
520 FOR J=12 TO 17
530 PRINT @14,J;CHR$(149);"      "
540 NEXT J
550 PRINT @4,20;CHR$(146);"THE FLUID IS
SAFELY DILUTED";CHR$(151)
560 SC=SC+1
570 WAIT 300
580 GOTO 80
590 CLS
600 PRINT @10,9;CHR$(149);"ALL LIVES LO
ST";CHR$(151)

```




**158** *Chemist*

```
610 PRINT @6,12;CHR$(149);"YOUR FINAL S  
CORE IS ";SC;CHR$(151)  
620 END
```

# 39

## Odd Word Out



At last - a chance to blast away with a laser gun all those words that don't fit.

You will be given a selection of words which either mean the same thing or are related in some way except for one 'odd man out' which must be identified and shot down.

### How to play

This is a game for two players so put in your names, remembering to press RETURN between each one.

Your screen will now display four words, for example:

herring, pike, bridge, roach

The odd man out here is bridge - the others are all fish.

It could have been herring, pike, roach, bream, however, and then which would have been the odd man out?

The faster you shoot down the wrong word the higher the points score you achieve. If you shoot down the wrong word however you get no points and the game returns to your partner.

Use any key to fire the gun.

### Programming Hints

Random selection may mean that the same sequence may pop up twice running.

If you would like to use your own selection of words then you can change the data in lines 800 to 970.

### Program

```

10 REM ODD WORD OUT
20 CLS
30 PAPER 7:INK 0
40 DIM W$(54)
60 PRINT CHR$(17)
90 FOR J=0 TO 15
100 READ N
110 POKE 46856+J,N
120 NEXT J
130 DATA 0,12,12,30,30,63,63,63
140 DATA 0,12,12,12,12,12,12,0
144 PRINT:PRINT
150 INPUT "1 OR 2 PLAYERS";N
152 IF N<1 OR N>2 THEN 150
160 PRINT @3,6;CHR$(146);"WHAT ARE THE
PLAYERS NAMES?";CHR$(151)
170 FOR J=1 TO N
180 PRINT @3,20;CHR$(146);"PLAYER ";J;"
?";CHR$(151)
184 PRINT @3,22;"
"
```

```

186 PRINT @3,22;" ";
190 INPUT P$(J)
200 PRINT @5,10+J*2;CHR$(147);P$(J);CHR
$(151)
210 NEXT J
220 WAIT 150
230 CLS
240 FOR J=1 TO 54
250 READ W$(J)
260 NEXT J
270 FOR T=1 TO N
280 PRINT @4,1;CHR$(147);P$(1);": ";S(1)
;CHR$(151)
284 IF N<2 THEN 294
290 PRINT @21,1;CHR$(147);P$(2);": ";S(2)
);CHR$(151)
294 IF S(1)>=100 OR S(2)>=100 THEN 770
300 G=INT(RND(1)*18)
310 G=G*3+1
314 PS=10
320 PRINT @6,6;CHR$(149);"YOUR TURN, ";
P$(T);CHR$(151)
324 WAIT 150
330 R1=INT(RND(1)*4)
340 R2=INT(RND(1)*4)
350 IF R2=R1 THEN 340
360 R3=INT(RND(1)*4)
370 IF R3=R2 OR R3=R1 THEN 360
380 R4=INT(RND(1)*4)
390 IF R4=R3 OR R4=R2 OR R4=R1 THEN 380
400 E=INT(RND(1)*54)+1
410 IF E<G+3 AND E>G-1 THEN 400
420 PRINT @R1*9+2,10;CHR$(144+(RND(1)*6
)+1);W$(G);
430 PRINT @R2*9+2,10;CHR$(144+(RND(1)*6
)+1);W$(G+1);
440 PRINT @R3*9+2,10;CHR$(144+(RND(1)*6
)+1);W$(G+2);
450 PRINT @R4*9+2,10;CHR$(144+(RND(1)*6
)+1);W$(E);
454 PRINT @ 38,10;CHR$(151)
460 P=1
470 D=1
480 PRINT @P,20;"  "
490 P=P+D

```

```

500 PRINT @P,20;CHR$(132);"a"
510 A$=KEY$
520 IF A$<>" " THEN 550
530 IF P>36 OR P<2 THEN D=-D:PS=PS-1
540 GOTO 480
550 ZAP
560 FOR J=18 TO 11 STEP -1
580 PRINT @P,J;CHR$(129);"b"
600 PRINT @P,J+1;" "
610 NEXT J
614 PRINT @P,J+1;" "
620 IF P>=R4*9+2 AND P<=R4*9+9 THEN 650


630 PING
640 GOTO 730
650 EXPLODE
660 FOR J=1 TO 10
670 PRINT @R4*9+2,10;CHR$(150);W$(E);CH
R$(151)
700 PRINT @R4*9+2,10;CHR$(147);W$(E);CH
R$(151)
710 NEXT J
716 IF PS<1 THEN PS=1
720 S(T)=S(T)+PS
722 PRINT @8,16;CHR$(145);"SCORE ";PS;"
POINTS!";CHR$(151)
730 FOR D=1 TO 800:NEXT D
740 CLS
750 NEXT T
760 GOTO 270
770 WAIT 100
782 IF S(2)>=100 THEN 786
784 PRINT @5,10;CHR$(149);P$(1);" IS TH
E WINNER!";CHR$(151):GOTO 788
786 PRINT @5,10;CHR$(149);P$(2);" IS TH
E WINNER!";CHR$(151)
788 PRINT @2,23;" "
790 END
800 DATA "LONDON","PARIS","ROME"
810 DATA "CAT","GERBIL","HORSE"
820 DATA "FORD","BRIDGE","FERRY"
830 DATA "ROACH","PIKE","HERRING"
840 DATA "QUARRY","QUIZ","QUARTER"
850 DATA "PING","BANG","CLICK"
860 DATA "DORSET","SUSSEX","SUFFOLK"
870 DATA "YACHT","DINGHY","SLOOP"

```

880 DATA "MINIM", "LEVEL", "REFER"  
890 DATA "FRANCE", "GERMANY", "ITALY"  
900 DATA "MAISON", "CHIEN", "FENETRE"  
910 DATA "SPECTRUM", "DRAGON", "ORIC"  
920 DATA "CRICKET", "FOOTBALL", "TENNIS"  
930 DATA "GEORGE", "CHARLES", "RICHARD"  
940 DATA "STOUR", "TRENT", "THAMES"  
950 DATA "DYFED", "GWENT", "CLWYD"  
960 DATA "CANARY", "FINCH", "SPARROW"  
970 DATA "DRAW", "POKE", "PRINT"

# 40

## Science Quiz



This is a program of multiple choice questions where you are given a selection of statements and asked to choose the correct answer.

If you find these questions too difficult or too easy you can ask your parents to provide your computer with some questions of your own. We will tell you how at the end of this introduction.

### How to play

The computer will ask for your name and you type it in and press RETURN

A choice of statements will appear on the screen and you will be asked to choose the correct answer.

Example: 'What will you grow if you plant an acorn?'

1. An elm tree
2. A rose
3. An oak tree

To answer press 1, 2 or 3

If you are correct the computer will tell you so.

If you are wrong the computer will tell you 'That is wrong'. Remember that the faster you answer, the more points you score. If you are too slow, you will run out of time.

### Programming Hints

The lines to change for your own questions are 600, to 790. Type in the question first, then the correct answer followed by two wrong answers. Remember to enclose each item in inverted commas (" ") as shown on the listing.

### Program

```

10 REM SCIENCE QUIZ
40 CLS
50 PRINT:PRINT
70 DIM Q$(20)
80 DIM A$(20,3)
90 PAPER 0:INK 4
100 INPUT "WHAT IS YOUR NAME";N$
110 FOR J=1 TO 20
120 READ Q$(J)
130 FOR K=1 TO 3
140 READ A$(J,K)
150 NEXT K
160 NEXT J
170 CLS
180 PRINT @3,1;CHR$(147);CHR$(128);"QUE
STIONS: ";QN;CHR$(144)

```



```

190 PRINT @22,1;CHR$(147);CHR$(128);"SC
ORE: ";SC;CHR$(144)
200 R=INT(RND(1)*20)+1
210 IF R=RS THEN 200
220 RS=R
230 PRINT @2,7;Q$(R)
240 RA=INT(RND(1)*3)+1
250 RB=INT(RND(1)*3)+1
260 IF RB=RA THEN 250
270 RC=INT(RND(1)*3)+1
280 IF RC=RA OR RC=RB THEN 270
290 PRINT @2,10;"1. ";A$(R,RA)
300 PRINT @2,12;"2. ";A$(R,RB)
310 PRINT @2,14;"3. ";A$(R,RC)
320 PRINT @2,18;CHR$(149);CHR$(128);"PR
ESS 1, 2 OR 3, ";N$;CHR$(144)
330 QN=QN+1
340 FOR T=1 TO 300
350 R$=KEY$
360 IF T/50<>INT(T/50) THEN 370
362 PLAY 1,0,0,0
364 MUSIC 1,3,T/50+1,8
366 WAIT 30
368 PLAY 0,0,0,0
370 IF R$="1" THEN T$=A$(R,RA):GOTO 440
380 IF R$="2" THEN T$=A$(R,RB):GOTO 440
390 IF R$="3" THEN T$=A$(R,RC):GOTO 440
400 NEXT T
410 EXPLODE
420 PRINT @2,20;CHR$(146);CHR$(128);"OU
T OF TIME!";CHR$(144)
430 GOTO 520
440 IF A$(R,1)=T$ THEN 480
450 EXPLODE
460 PRINT @2,20;CHR$(146);CHR$(128);"TH
AT IS WRONG, ";N$;CHR$(144)
470 GOTO 520
480 PING
490 PRINT @2,20;CHR$(146);CHR$(128);"CO
RRECT!";CHR$(144)
492 SI=10-INT(T/35)
500 SC=SC+SI
520 WAIT 300
522 IF QN=20 THEN 540
530 GOTO 170
540 CLS

```

```

542 PLAY 1,0,0,0:MUSIC 1,3,1,10:WAIT 30
546 MUSIC 1,3,5,10:WAIT 50
548 MUSIC 1,3,9,10:WAIT 90:PLAY 0,0,0,0

550 PRINT @4,10;CHR$(150);CHR$(128);"YO
UR FINAL SCORE IS ";SC;CHR$(151)
560 PAPER 7:INK 0
580 END
600 DATA "WHICH PLANET IS NEAREST TO TH
E SUN?","MERCURY","MARS","VENUS"
610 DATA "WHAT IS THE COMMON NAME FOR S
ODIUM CHLORIDE?","SALT"
615 DATA "AMMONIA","SUGAR"
620 DATA "WHICH WILL FLOAT IN WATER?","
CORK","IRON","GLASS"
630 DATA "WHAT WILL GROW IF YOU PLANT A
N ACORN?","AN OAK TREE","A ROSE"
635 DATA "AN ELM TREE"
640 DATA "AT WHAT TEMPERATURE WILL WATE
R FREEZE?","0 CENTIGRADE"
645 DATA "100 CENTIGRADE","-212 CENTIGR
ADE"
650 DATA "WHAT IS THE BOILING POINT OF
WATER?","100 CENTIGRADE"
655 DATA "0 CENTIGRADE","212 CENTIGRADE
"
660 DATA "WHICH WILL DISSOLVE IN WATER?
","SUGAR","SAND","WOOD"
670 DATA "WHICH IS POISONOUS?","CYANIDE
","GLUCOSE","SUCROSE"
680 DATA "WHAT DO BUTTERFLIES EAT?","NE
CTAR","CABBAGES","ALMOST ANYTHING"
690 DATA "IF YOU DROPPED AN IRON BALL A
ND A PEA,WHICH WOULD FALL FASTEST?"
695 DATA "NEITHER","THE IRON BALL","THE
PEA"
700 DATA "HOW MANY LEGS HAS A SPIDER?","
EIGHT","FOUR","SIX"
710 DATA "WHAT WILL YOU GET IF YOU MIX
BLUE AND YELLOW?","GREEN"
715 DATA "BLUE AND YELLOW STRIPES","ORA
NGE"
720 DATA "WHICH TREE DO CONKERS COME FR
OM?","HORSE CHESTNUT","OAK","BEECH"
730 DATA "HOW FAR WILL A CAR TRAVELLING
AT 60 KM/HR MOVE IN 15 MINUTES?"

```

735 DATA "15 KM", "60 KM", "30 KM"

740 DATA "WHICH WOULD FEEL HOTTEST IN A  
HOT DRINK?", "METAL SPOON"

745 DATA "PLASTIC SPOON", "PENCIL"

750 DATA "WHICH WILL CONDUCT ELECTRICIT  
Y?", "COPPER", "WOOD", "RUBBER"

760 DATA "A CYGNET IS A YOUNG", "SWAN", "  
DUCK", "GOOSE"

770 DATA "WHICH IS AN ELECTRICAL INSULA  
TOR?", "RUBBER", "COPPER", "IRON"

780 DATA "AN ECLIPSE OF THE SUN IS CAUS  
ED BY", "THE MOON", "A THUNDERSTORM"

785 DATA "A POWER CUT"

790 DATA "WHICH WILL BE ATTRACTED BY A  
MAGNET?", "IRON", "COPPER", "GOLD"

<b>APPLE II</b>				<b>COMMODORE 64 WARGAMING</b>				<b>MEMOTECH</b>				<b>SOFTWARE GUIDES</b>	
<b>APPLE II PROGRAMMER'S HANDBOOK</b>				0 246 12410 5	£6.95			<b>MEMOTECH COMPUTING</b>				<b>WORKING WITH dBASE II</b>	
0 246 12027 4	£10.95							0 246 12408 3	£5.95			0 246 12376 1	£7.95
<b>AQUARIUS</b>				<b>SOFTWARE 64: PRACTICAL PROGRAMS FOR THE COMMODORE 64</b>				<b>THE MEMOTECH GAMES BOOK</b>				<b>USING YOUR MICRO</b>	
<b>THE AQUARIUS AND HOW TO GET THE MOST FROM IT</b>				0 246 12266 8	£5.95			0 246 12407 5	£5.95			<b>COMPUTING FOR THE HOBBYIST AND SMALL BUSINESS</b>	
0 246 12295 1	£5.95			<b>INTRODUCING COMMODORE 64 MACHINE CODE</b>				<b>ORIC-1</b>				0 246 12023 1	£6.95
<b>ATARI</b>				0 246 12338 9	£7.95			<b>THE ORIC-1 AND HOW TO GET THE MOST FROM IT</b>				<b>DATABASES FOR FUN AND PROFIT</b>	
<b>GET MORE FROM THE ATARI</b>				<b>40 EDUCATIONAL GAMES FOR THE COMMODORE 64</b>				0 246 12130 0	£5.95			0 246 12032 0	£5.95
0 246 12149 1	£5.95			0 246 12318 4	£5.95			<b>THE ORIC PROGRAMMER</b>				<b>FIGURING OUT FACTS WITH A MICRO</b>	
<b>THE ATARI BOOK OF GAMES</b>				<b>DRAGON</b>				0 246 12157 2	£6.95			0 246 12221 8	£5.95
0 246 12277 3	£5.95			<b>THE DRAGON 32 AND HOW TO MAKE THE MOST OF IT</b>				<b>THE ORIC BOOK OF GAMES</b>				<b>INSIDE YOUR COMPUTER</b>	
<b>BBC MICRO</b>				0 246 12114 9	£5.95			0 246 12155 6	£5.95			0 246 12235 8	£4.95
<b>ADVANCED MACHINE CODE TECHNIQUES FOR THE BBC MICRO</b>				<b>THE DRAGON 32 BOOK OF GAMES</b>				<b>T199/4A</b>				<b>SIMPLE INTERFACING PROJECTS</b>	
0 246 12227 7	£6.95			0 246 12102 5	£5.95			<b>GET MORE FROM THE T199/4A</b>				0 246 12026 6	£6.95
<b>ADVANCED PROGRAMMING FOR THE BBC MICRO</b>				<b>THE DRAGON PROGRAMMER</b>				0 246 12281 1	£5.95			<b>PROGRAMMING</b>	
0 246 12158 0	£5.95			0 246 12133 5	£5.95			<b>VIC-20</b>				<b>COMPLETE GRAPHICS PROGRAMMER</b>	
<b>THE BBC MICRO: AN EXPERT GUIDE</b>				<b>DRAGON GRAPHICS AND SOUND</b>				<b>GET MORE FROM THE VIC-20</b>				0 246 12280 3	£6.95
0 246 12014 2	£6.95			0 246 12147 5	£6.95			0 246 12148 3	£5.95			<b>THE COMPLETE PROGRAMMER</b>	
<b>BBC MICRO GRAPHICS AND SOUND</b>				<b>INTRODUCING DRAGON MACHINE CODE</b>				<b>THE VIC-20 GAMES BOOK</b>				0 246 12015 0	£5.95
0 246 12156 4	£6.95			0 246 12324 9	£7.95			0 246 12187 4	£5.95			<b>PROGRAMMING WITH GRAPHICS</b>	
<b>DISCOVERING BBC MICRO MACHINE CODE</b>				<b>ELECTRON</b>				<b>ZX SPECTRUM</b>				0 246 12021 5	£5.95
0 246 12160 2	£6.95			<b>ADVANCED ELECTRON MACHINE CODE TECHNIQUES</b>				<b>AN EXPERT GUIDE TO THE SPECTRUM</b>				<b>WORD PROCESSING</b>	
<b>DISK SYSTEMS FOR THE BBC MICRO</b>				0 246 12403 2	£6.95			0 246 12278 1	£6.95			<b>CHOOSING A WORD PROCESSOR</b>	
0 246 12325 7	£7.95			<b>ADVANCED PROGRAMMING FOR THE ELECTRON</b>				<b>INTRODUCING SPECTRUM MACHINE CODE</b>				0 246 12347 8	£7.95
<b>HANDBOOK OF PROCEDURES AND FUNCTIONS FOR THE BBC MICRO</b>				0 246 12402 4	£5.95			0 246 12082 7	£7.95			<b>WORD PROCESSING FOR BEGINNERS</b>	
0 246 12415 6	£6.95			<b>ADVENTURE GAMES FOR THE ELECTRON</b>				<b>LEARNING IS FUN: 40 EDUCATIONAL GAMES FOR THE SPECTRUM</b>				0 246 12353 2	£5.95
<b>INTRODUCING THE BBC MICRO</b>				0 246 12417 2	£6.95			0 246 12233 1	£5.95			<b>FOR YOUNGER READERS</b>	
0 246 12146 7	£5.95			<b>ELECTRON GRAPHICS AND SOUND</b>				<b>MAKE THE MOST OF YOUR ZX MICRODRIVE</b>				<b>BEGINNERS' MICRO GUIDES: ZX SPECTRUM</b>	
<b>LEARNING IS FUN: 40 EDUCATIONAL GAMES FOR THE BBC MICRO</b>				0 246 12411 3	£6.95			0 246 12406 7	£5.95			0 246 12259 5	£2.95
0 246 12317 6	£5.95			<b>ELECTRON MACHINE CODE FOR BEGINNERS</b>				<b>THE SPECTRUM BOOK OF GAMES</b>				<b>BEGINNERS' MICRO GUIDES: BBC MICRO</b>	
<b>TAKE OFF WITH THE ELECTRON AND BBC MICRO</b>				0 246 12152 1	£7.95			0 246 12047 9	£5.95			0 246 12260 9	£2.95
0 246 12356 7	£5.95			<b>THE ELECTRON PROGRAMMER</b>				<b>SPECTRUM GRAPHICS AND SOUND</b>				<b>BEGINNERS' MICRO GUIDES: ACORN ELECTRON</b>	
<b>21 GAMES FOR THE BBC MICRO</b>				0 246 12340 0	£5.95			0 246 12192 0	£6.95			0 246 12381 8	£2.95
0 246 12103 3	£5.95			<b>HANDBOOK OF PROCEDURES AND FUNCTIONS FOR THE ELECTRON</b>				<b>THE SPECTRUM PROGRAMMER</b>				<b>MICROMATES</b>	
<b>PRACTICAL PROGRAMS FOR THE BBC MICRO</b>				0 246 12416 4	£6.95			0 246 12025 8	£5.95			<b>SIMPLE ANIMATION</b>	
0 246 12405 9	£6.95			<b>PRACTICAL PROGRAMS FOR THE ELECTRON</b>				<b>THE ZX SPECTRUM AND HOW TO GET THE MOST FROM IT</b>				0 246 12273 0	£1.95
<b>THE COLOUR GENIE</b>				0 246 12362 1	£7.95			<b>WHICH COMPUTER?</b>				<b>SIMPLE PICTURES</b>	
<b>MASTERING THE COLOUR GENIE</b>				<b>21 GAMES FOR THE ELECTRON</b>				<b>CHOOSING A MICROCOMPUTER</b>				0 246 12269 2	£1.95
0 246 12190 4	£5.95			0 246 12344 3	£5.95			0 246 12029 0	£4.95			<b>SIMPLE SHAPES</b>	
<b>COMMODORE 64</b>				<b>40 EDUCATIONAL GAMES FOR THE ELECTRON</b>				<b>LANGUAGES</b>				0 246 12271 4	£1.95
<b>BUSINESS SYSTEMS ON THE COMMODORE 64</b>				0 246 12404 0	£5.95			<b>COMPUTER LANGUAGES AND THEIR USES</b>				<b>SIMPLE SOUNDS</b>	
0 246 12422 9	£6.95			<b>TAKE OFF WITH THE ELECTRON AND BBC MICRO</b>				0 246 12022 3	£5.95			0 246 12270 6	£1.95
<b>ADVENTURE GAMES FOR THE COMMODORE 64</b>				0 246 12356 7	£5.95			<b>EXPLORING FORTH</b>				<b>SIMPLE SPELLING</b>	
0 246 12412 1	£6.95			<b>IBM</b>				0 246 12188 2	£5.95			0 246 12272 2	£1.95
<b>COMMODORE 64 COMPUTING</b>				<b>THE IBM PERSONAL COMPUTER</b>				<b>INTRODUCING LOGO</b>				<b>SIMPLE SUMS</b>	
0 246 12030 4	£5.95			0 246 12151 3	£6.95			0 246 12323 0	£5.95			0 246 12268 4	£1.95
<b>COMMODORE 64 DISK SYSTEMS AND PRINTERS</b>				<b>LYNX</b>				<b>INTRODUCING PASCAL</b>				<b>GRANADA GUIDES: COMPUTERS</b>	
0 246 12409 1	£6.95			<b>LYNX COMPUTING</b>				0 246 12322 2	£5.95			0 246 11895 4	£1.95
<b>THE COMMODORE 64 GAMES BOOK</b>				0 246 12131 9	£6.95			<b>MACHINE CODE</b>					
0 246 12258 7	£5.95			<b>8080 MACHINE CODE FOR HUMANS</b>				0 246 12031 2	£7.95				
<b>COMMODORE 64 GRAPHICS AND SOUND</b>				<b>6502 MACHINE CODE FOR HUMANS</b>				<b>Z80 MACHINE CODE FOR HUMANS</b>					
0 246 12342 7	£6.95			0 246 12076 2	£7.95			0 246 12076 2	£7.95				

If your family already owns an Oric Atmos – or if you are thinking of buying one – the chances are that it's because you have thought about the part computers are going to play in your family's future. The programs in this book have been designed to help the younger members of your family to handle the Oric Atmos and to increase their general knowledge – whilst enjoying themselves.

All of these ideas have been developed with the assistance of educationalists and a professional programmer. They are intended to give young users games which will complement their schooling and also give them a head start in using a micro.

Using a variety of 'games aids' such as beat-the-clock, stop-the-hangman, race-the-buzzer, subjects include languages, geography, mathematics and science. Hints are also provided to show how the contents of the programs can be changed to suit your family as their skill improves.

### *The Author*

Vince Apps is a regular contributor to journals such as *Popular Computing Weekly* and *Home Computing Weekly*. He is a graduate of Sussex University in Computer Science and has his own successful software company.

More books on the Oric and Atmos from Granada

#### **THE ATMOS PROGRAMMER**

*S. M. Gee and Mike James*

0 246 12535 7

#### **THE ATMOS BOOK OF GAMES**

*Mike James, S. M. Gee and Kay Ewbank*

0 246 12534 9

#### **ORIC AND ATMOS MACHINE CODE**

*Ian Sinclair*

0 246 12150 5

#### **THE ORIC PROGRAMMER**

*S. M. Gee and Mike James*

0 246 12157 2

#### **THE ORIC BOOK OF GAMES**

*Mike James, S. M. Gee and*

*Kay Ewbank*

0 246 12155 6

#### **THE ORIC-1 AND HOW TO GET THE MOST FROM IT**

*Ian Sinclair*

0 246 12130 0

Front cover illustration by Angus McKie

**GRANADA PUBLISHING**

Printed in Great Britain

0 246 12557 8

**£5.95 net**



APCS

40 EDUCATIONAL GRANTS FOR THE ORIGINOS GRANDA